
CAPI User Guide

Version 6.0



Copyright and Trademarks

LispWorks CAPI User Guide (Unix version)

Version 6.0

November 2009

Copyright © 2009 by LispWorks Ltd.

All Rights Reserved. No part of this publication may be reproduced, stored in a retrieval system, or transmitted, in any form or by any means, electronic, mechanical, photocopying, recording, or otherwise, without the prior written permission of LispWorks Ltd.

The information in this publication is provided for information only, is subject to change without notice, and should not be construed as a commitment by LispWorks Ltd. LispWorks Ltd assumes no responsibility or liability for any errors or inaccuracies that may appear in this publication. The software described in this book is furnished under license and may only be used or copied in accordance with the terms of that license.

LispWorks and KnowledgeWorks are registered trademarks of LispWorks Ltd.

Adobe and PostScript are registered trademarks of Adobe Systems Incorporated. Other brand or product names are the registered trademarks or trademarks of their respective holders.

The code for walker.lisp and compute-combination-points is excerpted with permission from PCL, Copyright © 1985, 1986, 1987, 1988 Xerox Corporation.

The XP Pretty Printer bears the following copyright notice, which applies to the parts of LispWorks derived therefrom:

Copyright © 1989 by the Massachusetts Institute of Technology, Cambridge, Massachusetts.

Permission to use, copy, modify, and distribute this software and its documentation for any purpose and without fee is hereby granted, provided that this copyright and permission notice appear in all copies and supporting documentation, and that the name of M.I.T. not be used in advertising or publicity pertaining to distribution of the software without specific, written prior permission. M.I.T. makes no representation about the suitability of this software for any purpose. It is provided "as is" without express or implied warranty. M.I.T. disclaims all warranties with regard to this software, including all implied warranties of merchantability and fitness. In no event shall M.I.T. be liable for any special, indirect or consequential damages or any damages whatsoever resulting from loss of use, data or profits, whether in an action of contract, negligence or other tortious action, arising out of or in connection with the use or performance of this software.

LispWorks contains part of ICU software obtained from <http://source.icu-project.org> and which bears the following copyright and permission notice:

ICU License - ICU 1.8.1 and later

COPYRIGHT AND PERMISSION NOTICE

Copyright © 1995-2006 International Business Machines Corporation and others. All rights reserved.

Permission is hereby granted, free of charge, to any person obtaining a copy of this software and associated documentation files (the "Software"), to deal in the Software without restriction, including without limitation the rights to use, copy, modify, merge, publish, distribute, and/or sell copies of the Software, and to permit persons to whom the Software is furnished to do so, provided that the above copyright notice(s) and this permission notice appear in all copies of the Software and that both the above copyright notice(s) and this permission notice appear in supporting documentation.

THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY, FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT OF THIRD PARTY RIGHTS. IN NO EVENT SHALL THE COPYRIGHT HOLDER OR HOLDERS INCLUDED IN THIS NOTICE BE LIABLE FOR ANY CLAIM, OR ANY SPECIAL INDIRECT OR CONSEQUENTIAL DAMAGES, OR ANY DAMAGES WHATSOEVER RESULTING FROM LOSS OF USE, DATA OR PROFITS, WHETHER IN AN ACTION OF CONTRACT, NEGLIGENCE OR OTHER TORTIOUS ACTION, ARISING OUT OF OR IN CONNECTION WITH THE USE OR PERFORMANCE OF THIS SOFTWARE.

Except as contained in this notice, the name of a copyright holder shall not be used in advertising or otherwise to promote the sale, use or other dealings in this Software without prior written authorization of the copyright holder. All trademarks and registered trademarks mentioned herein are the property of their respective owners.

US Government Restricted Rights

The LispWorks Software is a commercial computer software program developed at private expense and is provided with restricted rights. The LispWorks Software may not be used, reproduced, or disclosed by the Government except as set forth in the accompanying End User License Agreement and as provided in DFARS 227.7202-1(a), 227.7202-3(a) (1995), FAR 12.212(a)(1995), FAR 52.227-19, and/or FAR 52.227-14 Alt III, as applicable. Rights reserved under the copyright laws of the United States.

Address

LispWorks Ltd
St. John's Innovation Centre
Cowley Road
Cambridge
CB4 0WS
England

Telephone

From North America: 877 759 8839
(toll-free)
From elsewhere: +44 1223 421860

Fax

From North America: 617 812 8283
From elsewhere: +44 870 2206189

www.lispworks.com

Contents

Preface	vii
1 Introduction to the CAPI	1
What is the CAPI?	1
The history of the CAPI	2
The CAPI model	2
2 Getting Started	5
Using the CAPI package	6
Creating a window	6
Linking code into CAPI elements	8
3 General Considerations	11
The correct thread for CAPI operations	11
Properties of the host window system	12
Using Motif	14
4 Creating Common Windows	17
Generic properties	17
Specifying titles	20

	Displaying and entering text	22
	Stream panes	25
	Miscellaneous button elements	26
	Adding a toolbar to an interface	29
5	Choices	31
	Choice classes	32
	List panels	34
	Graph panes	39
	Option panes	42
	Text input choice	43
	Menu components	43
	General properties of choices	43
6	Laying Out CAPI Panes	47
	Organizing panes in columns and rows	48
	Other types of layout	51
	Combining different layouts	52
	Constraining the size of layouts	54
	Advanced pane layouts	59
7	Modifying CAPI Windows	65
	Initialization	65
	Resizing	66
	Scrolling	67
	Swapping panes and layouts	67
	Setting the data in a pane	68
	Iconifying and restoring windows	68
	Destroying	69
8	Creating Menus	71
	Creating a menu	71
	Grouping menu items together	72
	Creating individual menu items	75
	The CAPI menu hierarchy	76
	Mnemonics in menus	78
	Alternative menu items	79

Disabling menu items 80

Menus with images 81

Popup menus for panes 82

9 Defining Interface Classes 85

The define-interface macro 85

An example interface 86

Adapting the example 88

Connecting an interface to an application 94

Controlling the interface title 97

10 Prompting for Input 99

Some simple dialogs 100

Prompting for values 101

Window-modal Cocoa dialogs 108

Dialog Owners 109

Creating your own dialogs 110

In-place completion 115

11 Creating Your Own Panes 117

Displaying graphics 117

Receiving input from the user 120

Creating graphical objects 122

12 Graphics Ports 131

Introduction 131

Features 132

Graphics state 133

Graphics state transforms 133

Graphics state operation 135

Pixmap graphics ports 136

Portable font descriptions 136

Working with images 138

13 The Color System 145

Introduction 145

Reading the color database 146

- Color specs 147
- Color aliases 148
- Color models 149
- Loading the color database 151
- Defining new color models 151

14 Printing from the CAPI—the Hardcopy API 153

- Printers 153
- Printer definition files 154
- PPD files 154
- Print jobs 155
- Handling pages—page on demand printing 155
- Handling pages—page sequential printing 155
- Printing a page 155
- Other printing functions 156

Index 157

Preface

This preface contains information you need when using the rest of the CAPI documentation. It discusses the purpose of this manual, the typographical conventions used, and gives a brief description of the rest of the contents.

Assumptions

The CAPI documentation assumes that you are familiar with:

- LispWorks
- Common Lisp and CLOS, the Common Lisp Object System
- UNIX, Linux or FreeBSD, and the X Window System with GTK+ or Motif

Illustrations in this manual show the CAPI running on Linux under GTK+ with a particular window manager and theme, so if you use GTK+ with a different window manager or theme, or Motif, you should expect some variation from the figures depicted here.

Unless otherwise stated, examples given in this document assume that the current package has `CAPI` on its package-use-list.

Conventions used in the manual

Throughout this manual, certain typographical conventions have been adopted to aid readability.

1. Whenever an instruction is given, is numbered and printed like this.

Text which you should enter explicitly is printed `like this`.

A Description of the Contents

This guide forms an introductory course in developing applications using the CAPI. Please note that, like the rest of the LispWorks documentation, it does assume knowledge of Common Lisp.

Chapter 1, *Introduction to the CAPI*, introduces the principles behind the CAPI, some of its fundamental concepts, and what it sets out to achieve.

Chapter 2, *Getting Started*, presents a series of simple examples whose aim is to familiarize you with some of the most important elements and functions.

Chapter 3, *General Considerations*, covers some general issues that you should be aware of when using CAPI, including information about the host window system.

Chapter 4, *Creating Common Windows*, introduces more of the fundamental CAPI elements and common themes. These elements are explained in greater detail in the remainder of the manual.

Chapter 5, *Choices*, explains the key CAPI concept of the *choice*. A choice groups CLOS objects together and provides the notion of there being a selected object amongst that group of objects. Button panels and list panels are examples of choices.

Chapter 6, *Laying Out CAPI Panes* introduces the idea of *layouts*. These let you combine different CAPI elements inside a single window.

Chapter 7, *Modifying CAPI Windows*, outlines basic techniques for modifying existing windows.

Chapter 8, *Creating Menus*, shows you how to add menus to a window.

Chapter 9, *Defining Interface Classes*, introduces the macro `define-interface`. This macro can be used to define interface classes composed of CAPI elements

— either the predefined elements explained elsewhere in this manual or your own.

Chapter 10, *Prompting for Input*, discusses the ways in which dialog boxes may be used to prompt the user for input.

Chapter 11, *Creating Your Own Panes*, shows you how you can define your own classes when those provided by the CAPI are not sufficient for your needs.

Chapter 12, *Graphics Ports*, provides information on the Graphics Ports package, which provides a selection of drawing and image transformation functions. Although not part of the CAPI package, and therefore not strictly part of the CAPI, the Graphics Ports functions are used in conjunction with CAPI panes, and are therefore documented in this manual and the *LispWorks CAPI Reference Manual*.

Chapter 13, *The Color System*, allows applications to use keyword symbols as aliases for colors in Graphics Ports drawing functions. They can also be used for backgrounds and foregrounds of windows and CAPI objects.

Chapter 14, *Printing from the CAPI—the Hardcopy API*, describes the programmatic printing of Graphics Ports.

The Reference Manual

The second part of the CAPI documentation is the *LispWorks CAPI Reference Manual*. This provides a complete description of every CAPI class, function and macro, and also provides a reference chapter on the Graphics Port functions. Entries are listed alphabetically, and the typographical conventions used are similar to those used in *Common Lisp: the Language* (2nd Edition) (Steele, 1990).

1

Introduction to the CAPI

1.1 What is the CAPI?

The CAPI (Common Application Programmer's Interface) is a library for implementing portable window-based application interfaces. It is a conceptually simple, CLOS-based model of interface elements and their interaction. It provides a standard set of these elements and their behaviors, as well as giving you the opportunity to define elements of your own.

The CAPI's model of window-based user interfaces is an abstraction of the concepts that are shared between all contemporary window systems, such that you do not need to consider the details of a particular system. These hidden details are taken care of by a back end library written for that system alone.

An advantage of making this abstraction is that each of the system-specific libraries can be highly specialized, concentrating on getting things right for that particular window system. Furthermore, because the implementation libraries and the CAPI model are completely separate, libraries can be written for new window systems without affecting either the CAPI model or the applications you have written with it.

The CAPI currently runs under X Window System with either GTK+ or Motif, Microsoft Windows and Mac OS X. Using CAPI with Motif is deprecated.

1.2 The history of the CAPI

Window-based applications written with LispWorks 3 and previous used CLX², CLUE, and the LispWorks Toolkit. Such applications are restricted to running under X Windows. Because we and our customers wanted a way to write portable window code, we developed a new system for this purpose: the CAPI.

Part of this portability exercise was undertaken before the development of the CAPI, for graphics ports, the generic graphics library. This includes the portable color, font, and image systems in LispWorks. The CAPI is built on top of this technology, and has been implemented for Motif, Microsoft Windows, Cocoa and GTK+.

All Lisp-based environment and application development in LispWorks Ltd now uses the CAPI. We recommend that you use the CAPI for window-based application development in preference to the systems mentioned earlier.

1.3 The CAPI model

The CAPI provides an abstract hierarchy of classes which represent different sorts of window interface elements, along with functions for interacting with them. Instances of these classes represent window objects in an application, with their slots representing different aspects of the object, such as the text on a button, or the items on a menu. These instances are not actual window objects but provide a convenient representation of them for you. When you ask the CAPI to display your object, it creates a real window system object to represent it. This means that if you display a CAPI button, a real Windows button is created for it when running on Microsoft Windows, a real GTK+ button when running on GTK+, and a real Cocoa button when running on Cocoa.

A different approach would have been to simulate the window objects and their look and feel. This approach is problematic. Because the library makes itself entirely responsible for the application's look and feel, it may not simulate it correctly in obscure cases. Also, manufacturers occasionally change the look and feel of their window systems. Applications written with a library that simulates window objects will continue to have the old look and feel until the application is recompiled with an updated library.

The CAPI's approach makes the production of the screen objects the responsibility of the native window system, so it always produces the correct look and feel. Furthermore, the CAPI's use of the real interface to the window system means that it does not need to be upgraded to account for look and feel changes, and anything written with it is upwardly compatible, just like any well-written application.

1.3.1 CAPI Classes

There are four basic objects in the CAPI model: *interfaces*, *menus*, *panes* and *layouts*.

Everything that the CAPI displays is contained within an interface (an instance of the class `interface`). When an interface is displayed a window appears containing all the menus and panes you have specified for it.

An interface can contain a number of menus which are collected together on a menu bar. Each menu on the menu bar can contain menu items or other menus. Items can be grouped together visually and functionally inside *menu components*. Menus, menu items, and menu components are, respectively, instances of the classes `menu`, `menu-item`, and `menu-component`.

Panes are window objects such as buttons and lists. They can be positioned anywhere in an interface. The CAPI provides many different kinds of pane class, among them `push-button`, `list-panel`, `editor-pane`, `tree-view` and `graph-pane`.

The positions of panes are controlled by a layout, which allows objects to be collected together and positioned either regularly (with instances of the classes `column-layout`, `row-layout` or `grid-layout`) or arbitrarily using a `pinboard-layout`. Layouts themselves can be laid out by other layouts — for example, a row of buttons can be laid out above a list by placing both the `row-layout` and the list in a `column-layout`.

2

Getting Started

This chapter introduces some of the most basic CAPI elements and functions. The intention is simply that you should become familiar with the most useful elements available, before learning how you can use them constructively. You should work through the examples in this chapter.

A CAPI application consists of a hierarchy of CAPI objects. CAPI objects are created using `make-instance`, and although they are standard CLOS objects, CAPI slots should generally be accessed using the documented accessors, and not using the CLOS `slot-value` function. You should not rely on `slot-value` because the implementation of the CAPI classes may evolve.

Once an instance of a CAPI object has been created in an interface, it can be displayed on your screen using the function `display`.

2.1 Using the CAPI package

All symbols in this manual are exported from either the CAPI or COMMON-LISP packages unless explicitly stated otherwise. To access CAPI symbols, you could qualify them all explicitly in your code, for example `capi:output-pane`.

However it is more convenient to create a package which has CAPI on its package-use-list:

```
(defpackage "MY-PACKAGE"
  (:add-use-defaults t)
  (:use "CAPI")
)
```

This creates a package in which all the CAPI symbols are accessible. To run the examples in this guide, first evaluate

```
(in-package "MY-PACKAGE")
```

2.2 Creating a window

This section shows how easy it is to create a simple window, and how to include CAPI elements, such as panes, in your window.

1. Enter the following in a listener

```
(make-instance 'interface
  :visible-min-width 200
  :title "My Interface")

(display *)
```

Figure 2.1 Creating a simple window



A small window appears on your screen, called "My Interface". This is the most simple type of window that can be created with the CAPI.

Note: By default this window has a menu bar with the **Works** menu. The **Works** menu gives you access to a variety of LispWorks tools, just like the

Works menu of any window in the LispWorks IDE. It is automatically provided by default for any interface you create. You can omit it by passing `:auto-menus nil`.

The usual way to display an instance of a CAPI window is `display`. However, another function, `contain`, is provided to help you during the course of development.

Notice that the "My Interface" window cannot be made smaller than the minimum width specified. All CAPI geometry values (window size and position) are integers and represent pixel values.

Only a top level CAPI element is shown by `display` — that is, an instance of an `interface`. To display other CAPI elements (for example, buttons, editor panes, and so on), you must provide information about how they are to be arranged in the window. Such an arrangement is called a *layout* — you will learn more about layouts in Chapter 6.

On the other hand, `contain` automatically provides a default layout for any CAPI element you specify, and subsequently displays it. During development, it can be useful for displaying individual elements of interest on your screen, without having to create an interface for them explicitly. However, `contain` is only provided as a development tool, and should not be used for the final implementation of a CAPI element. See Chapter 9, "Defining Interface Classes" on how to display CAPI elements in an interface.

Note that a displayed CAPI element should only be accessed in its own thread. See "The correct thread for CAPI operations" on page 11 for more information about this.


This is how you can create and display a button using `contain`.

1. Enter the following into a listener:

```
(make-instance 'push-button
              :data "Button")
```

```
(contain *)
```

Figure 2.2 Creating a push-button interface



This creates an interface which contains a single push-button, with a label specified by the `:data` keyword. Notice that you could have performed the same example using `display`, but you would also have had to create a layout so that the button could have been placed in an interface and displayed.

You can click on the button, and it will respond in the way you would expect (it will depress). However, no code will be run which performs an action associated with the button. How to link code to window items is the topic of the next section.

2.3 Linking code into CAPI elements

Getting a CAPI element to perform an action is done by specifying a *callback*. This is a function which is performed whenever you change the state of a CAPI element. It calls a piece of code whenever a choice is made in a window.

Note that the result of the callback function is ignored, and that its usefulness is in its side-effects.

1. Try the following:

```
(make-instance 'push-button
               :data "Hello"
               :callback
               #'(lambda (&rest args)
                  (display-message
                   "Hello World"))))
```

(contain *)

Figure 2.3 Specifying a callback



2. Click on the **Hello** button.

A dialog appears containing the message “Hello World”.

Figure 2.4 A dialog displayed by a callback.



The CAPI provides the function `display-message` to allow you to pop up a dialog box containing a message and a Confirm button. This is one of many pre-defined facilities that the CAPI offers.

Note: When you develop CAPI applications, your application windows are run in the same Window system event loop as the LispWorks IDE. This - and the fact that in Common Lisp user code exists in the same global namespace as the Common Lisp implementation - means that a CAPI application running in the LispWorks IDE can modify the same values as you can concurrently modify from one of the the LispWorks IDE programming tools.

For example, your CAPI application might have a button that, when pressed, sets a slot in a particular object that you could also set by hand in the Listener.

Such introspection can be useful but can also lead to unexpected values and behavior while testing your application code.

3

General Considerations

This chapter describes general issues relating to the use of CAPI. Subsequent chapters address the use of particular CAPI elements.

3.1 The correct thread for CAPI operations

All operations on displayed CAPI elements need to be in the thread (that is, the `mp:process`) that runs their interface. On some platforms, `display` and `contain` make a new thread. On Cocoa, all interfaces run in a single thread.

In most cases this issue does not arise, because CAPI callbacks are run in the correct thread. However, if your code needs to communicate with a CAPI window from a random thread, it should use `execute-with-interface`, `execute-with-interface-if-alive` or `apply-in-pane-process` to send the function to the correct thread.

This is why the brief interactive examples in this manual generally use `execute-with-interface` or `apply-in-pane-process` when modifying a displayed CAPI element. In contrast, the demo example in “Connecting an interface to an application” on page 94 is modified only by callbacks which run in the demo interface’s own process, and so there is no need to use `execute-with-interface` or `apply-in-pane-process`.

3.2 Properties of the host window system

This section describes properties of the host window system that affect the appearance and behavior of CAPI windows.

3.2.1 Using Windows themes

On Microsoft Windows XP, Vista and Windows 7 LispWorks is *themed*. That is, it uses the current theme of the desktop.

It is possible to switch this off by calling the function `win32:set-application-themed` with argument `nil`.

`win32:set-application-themed` affects only windows that are created after it was called. Normally, it should be called before any window is created, so that all LispWorks windows will have a consistent appearance.

3.2.2 Matching resources

You can configure the LispWorks IDE and your application to use resources on GTK+ and Motif. The applicable resources determine the default fonts, colors and certain other properties used in CAPI elements.

The `element` initarg `widget-name` is used to match resources. CAPI gives a name for the main widget that it creates for each element that has a representation in the library. This name is then included in the "path" that GTK+ and Motif use to match resources for each widget.

3.2.2.1 Matching resources on GTK+

By default, the name of the widget is the name of the class of the element, downcased (except top level interfaces on GTK+, see the GTK+ specific part). You can override the name by either passing `widget-name` when making the element, or by calling `(setf element-widget-name)` before displaying the element.

To make it easier to define resources specific to the application, the CAPI GTK+ library, when using the default name, prepends the *application-class* (see `convert-to-screen`) followed by a dot. So for an interface of class `my-inter-`

face which is displayed in a screen with *application-class* "my-application", the default *widget-name* is:

```
my-application.my-interface
```

Example GTK+ resource files are in `examples/gtk/`.

3.2.2.2 Matching resources on X11/Motif

widget-name is used as described for GTK+ in “Matching resources on GTK+” on page 12, except that the default *widget-name* for a top level interface does include the prepended *application-class*.

The file `app-defaults/Lispworks`, supplied in the LispWorks library for relevant platforms, contains the application fallback resources for LispWorks 6.0 and illustrates resources you may wish to change.

The files `app-defaults/*-classic` contain the fallback resources that were supplied with LispWorks 4.4.

For further information about X resources, consult documentation for the X Window system.

3.2.2.3 Resources for LispWorks CAPI applications

Delivered applications which need fallback resources should pass the `:application-class` and `:fallback-resources` keys described in the *LispWorks CAPI Reference Manual* under `convert-to-screen`.

There is an example showing how to make a CAPI GUI configurable by GTK+ resources in `examples/capi/elements/gtk-resources.lisp`. To construct custom resources for your CAPI/GTK+ application, see the example resource files in `examples/gtk/`.

To construct custom X resources for your CAPI/Motif application, consult `app-defaults/Lispworks` which illustrates resources you may wish to change in your application.

3.2.2.4 X resources for in-place completion windows

The special window described in “In-place completion” on page 115 has interface with name `"non-focus-list-prompter"`. This name can be used to define

resources specific to the in-place completion window. The completion list is a `list-panel` and the filter is a `text-input-pane`.

3.2.3 The break gesture

If a CAPI window is busy and unresponsive you can use the break gesture `Meta+Ctrl+C` to regain control.

Note that this break gesture is specific to the window system your CAPI program is running in.

On GTK+ and Motif you can configure the break gesture by calling the function `set-interactive-break-gestures`.

3.3 Using Motif

This section describes how to use the Motif window system on supported platforms.

3.3.1 Using Motif on Linux, FreeBSD and x86/x64 Solaris

Use of Motif with LispWorks is deprecated on these platforms, but you can still use it.

LispWorks uses GTK+ as the default window system for CAPI and the LispWorks IDE on Linux, FreeBSD and x86/x64 Solaris.

To use Motif instead you need to load it explicitly, by:

```
(require "capi-motif")
```

Requiring the `"capi-motif"` module makes CAPI use Motif as its default library.

You can override the default library by specifying the appropriate CAPI screen. For more information about this, see the `screen` argument to `display` and `convert-to-screen`.

3.3.2 Using Motif on Macintosh

Use of Motif with LispWorks is deprecated on the Macintosh, but you can still use it.

LispWorks is supplied as two images. One uses Cocoa as the default window system for CAPI and the LispWorks IDE, the other uses GTK+ as its default window system. Only this latter image can use the alternative Motif window system.

To use Motif you need to load it into the GTK+ LispWorks image, by:

```
(require "capi-motif")
```

Requiring the "capi-motif" module makes CAPI use Motif as its default library.

You can override the default library by specifying the appropriate CAPI screen. For more information about this, see the *screen* argument to `display` and `convert-to-screen`.

Note: you cannot load Motif into the Cocoa image.

Note: the GTK+ LispWorks image is installed on Macintosh when you select the X11 GUI option at install time. See the *LispWorks Release Notes and Installation Guide* for further information on installing this option.

3.3.3 Using Motif on SPARC Solaris and HP-UX

LispWorks on SPARC Solaris and HP-UX does not support GTK+, and Motif is the only supported window system. You do not need to load it or specify the *screen* explicitly on these platforms.

3 *General Considerations*

4

Creating Common Windows

So far you have only seen two types of CAPI element: the interface (which is the top level CAPI element, and is present in any CAPI window) and the `push-button`. This section shows how you can use the CAPI to create other common windowing elements you are likely to need.

Before trying out the examples in this chapter, define the functions `test-callback` and `hello` in your Listener. The first displays the list of arguments it is given, and returns `nil`. The second just displays a message.

```
(defun test-callback (data interface)
  (display-message "Data ~S in interface ~S"
                  data interface))

(defun hello (data interface)
  (declare (ignore data interface))
  (display-message "Hello World"))
```

We will use these callbacks in future examples.

4.1 Generic properties

Because CAPI elements are just like CLOS classes, many elements share a common set of properties. This section describes the properties that all the classes described in this chapter inherit.

4.1.1 Scroll bars

The CAPI lets you specify horizontal or vertical scroll bars for any subclass of the `simple-pane` element (including all of the classes described in this chapter).

Horizontal and vertical scroll bars can be specified using the keywords `:horizontal-scroll` and `:vertical-scroll`. By default, both `:vertical-scroll` and `:horizontal-scroll` are `nil`.

4.1.2 Background and foreground colors

All subclasses of the simple pane element can have different foreground and background colors, using the `:background` and `:foreground` keywords. For example, including

```
:background :blue  
:foreground :yellow
```

in the `make-instance` of a text pane would result in a pane with a blue background and yellow text.

4.1.3 Fonts

The CAPI interface supports the use of other fonts for text in title panes and other CAPI objects, such as buttons, through the use of the `:font` keyword. If the CAPI cannot find the specified font it reverts to the default font. The `:font` keyword applies to data following the `:text` keyword. The value is a graphics ports `gp:font-description` object specifying various attributes of the font.

On systems running X Windows, the `xlsfonts` command can be used to list which fonts are available. The X logical font descriptor can be explicitly passed as a string to the `:font` initarg, which will convert them.

Here is an example of a `title-pane` with an explicit font:

```
(contain
  (make-instance 'title-pane
    :text "A title pane"
    :font (gp:make-font-description
      :family "times"
      :size 12
      :weight :medium
      :slant :roman)))
```

Here is an example of using `:font` to produce a title pane with larger lettering. Note that the CAPI automatically resized the pane to fit around the text.

```
(contain
  (make-instance 'title-pane
    :text "A large piece of text"
    :font (gp:make-font-description
      :family "times"
      :size 34
      :weight :medium
      :slant :roman)))
```

Figure 4.1 An example of the use of font descriptions



4.1.4 Mnemonics

This section applies to Windows and GTK+ only.

It is also possible to make mnemonics work on Motif, but not straightforward. Contact Lisp Support if you need help with that.

Underlined letters in menus, titles and buttons are called mnemonics. The user can select the element by pressing the corresponding key.

4.1.4.1 Controlling Mnemonics

For individual buttons, menus, menu items and title panes, you can use the `:mnemonic` initarg to control them. For example:

```
(capi:contain (make-instance 'capi:push-button
                             :data "FooBar"
                             :mnemonic #\B))
```

For more information on mnemonics in buttons, see “Mnemonics in buttons” on page 28 and the *LispWorks CAPI Reference Manual*.

For information on controlling mnemonics in button panels, see “Mnemonics in button panels” on page 34. For information on controlling mnemonics in menus, see “Mnemonics in menus” on page 78.

The initarg `:mnemonic-title` allows you to specify the mnemonic in the title for many pane classes including `list-panel`, `text-input-pane` and `option-pane`. Also `grid-layout` supports `mnemonic-title` when `has-title-column-p` is true. For the details see `titled-object` in the *LispWorks CAPI Reference Manual*.

4.1.4.2 Mnemonics on Microsoft Windows

On Microsoft Windows the user can make the mnemonics visible by holding down the `Alt` key.

Windows XP can hide mnemonics when the user is not using the keyboard. This is controlled by

Control Panel > Display > Appearance > Effects > Hide underlined letters...

4.2 Specifying titles

It is possible to specify a title for a window, or part of a window. Several of the examples that you have already seen have used titles. There are two ways that you can create titles: by using the `title-pane` class, or by specifying a title directly to any subclass of `titled-object`.

4.2.1 Title panes

A title pane is a blank pane into which text can be placed in order to form a title.

```
(setq title (make-instance 'title-pane
                          :visible-min-width 200
                          :text "Title"))

(contain title)
```

Figure 4.2 A title pane



4.2.2 Specifying titles directly

You can specify a title directly to all CAPI panes, using the `:title` keyword. This is much easier than using title-panes, since it does not necessitate using a layout to group two elements together.

Any class that is a subclass of `title-object` supports the `:title` keyword. All of the standard CAPI panes inherit from this class. You can find all the subclasses of `title-object` by graphing them using the class browser.

4.2.2.1 Window titles

Specify a title for a CAPI window by supplying the `:title` initarg for the `interface`, and access it with `interface-title`.

Further control over the title of your application windows can be achieved by using `set-default-interface-prefix-suffix` and/or specializing `interface-extend-title` as illustrated in “Controlling the interface title” on page 97.

4.2.2.2 Titles for elements

The position of any title can be specified by using the `:title-position` keyword. Most panes default their *title-position* to `:top`, although some use `:left`.

You can place the title in a frame (like a groupbox) around its element by specifying `:title-position :frame`.

You may specify the font used in the title via the keyword `:title-font`.

The title of a `titled-object`, and its font, may be changed interactively with the use of `setf`, if you wish.

1. Create a push button by evaluating the code below:

```
(setq button (make-instance 'push-button
                            :text "Hello"
                            :title "Press: "
                            :title-position :left
                            :callback 'hello))

(contain button)
```

2. Now evaluate the following:

```
(apply-in-pane-process
 button #'(setf titled-object-title) "Press here: " button)
```

As soon as the form is evaluated, the title of the pane you just created changes.

3. Lastly evaluate the following:

```
(apply-in-pane-process
 button #'(setf titled-object-title-font)
 (gp:merge-font-descriptions
 (gp:make-font-description :size 42)
 (gp:convert-to-font-description
 button
 (titled-object-title-font button)))) button)
```

Notice how the window automatically resizes in steps 2 and 3, to make allowance for the new size of the title.

4.3 Displaying and entering text

There are a variety of ways in which an application can display text, accept text input or allow editing of text by the user. Display panes show non-edit-

able text, text input panes are used for entering short pieces of text, and editor panes are commonly used for dealing with large amounts of text such as files. Rich text panes are available on Cocoa and Windows, supporting formatted text.

4.3.1 Display panes


Display panes can be used to display text messages on the screen. The text in these messages cannot be edited, so they can be used by the application to present a message to the user. The `:text` keyword can be used to specify the message that is to appear in the pane.

1. Create a display pane by evaluating the code below:

```
(setq display (make-instance 'display-pane
                             :text "This is a message"))

(contain display)
```

Figure 4.3 A display pane



Note that the window title, which defaults to "Container" for windows created by `contain`, may appear truncated.

4.3.2 Text input panes

When you want the user to enter a line of text — for instance a search string — a text input pane can be used.

```
(setq text (make-instance 'text-input-pane
                          :title "Search: "
                          :callback 'test-callback))

(contain text)
```

Figure 4.4 A text input pane



Notice that the default title position for text input panes is `:left`.

You can place text programmatically in the text input pane by supplying a string for the `:text` initarg, or later by calling `(setf text-input-pane-text)` in the appropriate process.

You can add toolbar buttons for easier user input via the `:buttons` initarg. This example allows the user to enter the filename of an existing Lisp source file, either directly or by selecting the file in a dialog raised by the **Browse File** button. There is also a **Cancel** button, but the default **OK** button is not displayed:

```
(capi:contain
 (make-instance
  'capi:text-input-pane
  :buttons
  (list :cancel t
        :ok nil
        :browse-file
        (list :operation :open
              :filter "*.LISP;*.LSP"))))
```

For a larger quantity of text use `multi-line-text-input-pane`.

4.3.3 Editor panes


Editor panes can be created using the `editor-pane` element.

```
(setq editor
  (make-instance 'editor-pane
    :text
    "some text in an editor pane"))

(contain editor)
```

The Editor tool in the LispWorks IDE, as described in the *LispWorks IDE User Guide* and the *LispWorks Editor User Guide*, uses `editor-pane`.

Figure 4.5 An editor pane



Note: when you supply the `:buffer-name` initarg and/or the `:text` initarg with positive length, then the `editor-pane` initially displays a new buffer containing that text and/or with the specified buffer name. If you do not supply one of those arguments, then the `editor-pane` displays some existing editor buffer chosen at random. See the *LispWorks CAPI Reference Manual* for details.

The cursor in an `editor-pane` blinks on and off under the control of the `editor-pane-blink-rate` mechanism.

An `editor-pane` can be made non-editable by users with the initarg `:enabled :read-only`, or completely disabled with `:enabled nil`.

4.4 Stream panes

There are three subclasses of `editor-pane` which handle Common Lisp streams.

4.4.1 Collector panes

A collector pane displays anything printed to the stream associated with it. Background output windows, for instance, are examples of collector panes.

1.

```
(contain (make-instance 'collector-pane
                        :title "Example collector pane:"))
```
2.

```
(princ "abc" (collector-pane-stream *))
```

4.4.2 Interactive streams

An interactive stream is the building block on which `listener-pane` is built.

```
(contain (make-instance 'interactive-stream
                        :title "Stream:"))
```

4.4.3 Listener panes

The `listener-pane` class is a subclass of `interactive-stream`, and allows you to create interactive Common Lisp sessions. You may occasionally want to include a listener pane in a tool (as, for instance, in the LispWorks IDE Debugger).

```
(contain (make-instance 'listener-pane
                        :title "Listener:"))
```

4.5 Miscellaneous button elements

A variety of different buttons can be created for use in an application. These include push buttons, which you have already seen, and check buttons. Button panels can also be created, and are described in Chapter 5, “Choices”.

4.5.1 Push buttons

You have already seen push buttons in earlier examples. The `:enabled` keyword can be used to specify whether or not the button should be selectable when it is displayed. This can be useful for disabling a button in certain situations.

The following code creates a push button which cannot be selected.

```
(setq offbutton (make-instance 'push-button
                              :data "Button"
                              :enabled nil))

(contain offbutton)
```

These `setf` expansions enable and disable the button:

```
(apply-in-pane-process
 offbutton #'(setf button-enabled) t offbutton)

(apply-in-pane-process
 offbutton #'(setf button-enabled) nil offbutton)
```

All subclasses of the `button` class can be disabled in this way.

4.5.2 Check buttons


Check buttons can be produced with the `check-button` element.

1. Enter the following in a Listener:

```
(setq check (make-instance 'check-button
                          :selection-callback 'hello
                          :retract-callback 'test-callback
                          :text "Button"))

(contain check)
```

Figure 4.6 A check button



Notice the use of `:retract-callback` in the example above, to specify a callback when the element is deselected.

Like push buttons, check buttons can be disabled by specifying `:enabled nil`.

4.5.3 Radio buttons

Radio buttons can be created explicitly although they are usually part of a button panel as described in Chapter 5, *Choices*. The `:selected` keyword is used to specify whether or not the button is selected, and the `:text` keyword can be used to label the button.

```
(contain (make-instance 'radio-button
                        :text "Radio Button"
                        :selected t))
```

Figure 4.7 An explicitly created radio button



Although a single radio button is of limited use, having an explicit radio button class gives you greater flexibility, since associated radio buttons need not be physically grouped together. Generally, the easiest way of creating a group of radio buttons is by using a button panel, but doing so means that they will be geometrically, as well as semantically, connected.

4.5.4 Mnemonics in buttons

The initarg `:mnemonic` allows you to supply a character, integer or symbol specifying a mnemonic for a button.

Alternatively you can specify the button text and its mnemonic together with the initarg `:mnemonic-text`, for example:

```
(contain
  (make-instance 'radio-button
                 :mnemonic-text
                 "Radio Button with a &Mnemonic"))
```

4.6 Adding a toolbar to an interface

Top level interfaces can have a toolbar, which is typically displayed at the top of the window. On Cocoa, this will be a standard foldable toolbar.

The end user can raise a customization dialog to choose which items appear on the toolbar. See the *toolbar-items* and *toolbar-states* initargs for `interface` and the functions `interface-toolbar-state`, `interface-default-toolbar-states`, `interface-update-toolbar` and `interface-customize-toolbar`.

4 *Creating Common Windows*

5

Choices

Some elements of a window interface contain collections of items, for example rows of buttons, lists of filenames, and groups of menu items. Such elements are known in the CAPI as *collections*.

In most collections, items may be selected by the user — for example, a row of buttons. Collections whose items can be selected are known as *choices*. Each button in a row of buttons is either checked or unchecked, showing something about the application's state — perhaps that color graphics are switched on and sound is switched off. This selection state came about as the result of a *choice* the user made when running the application, or default choices made by the application itself.

The CAPI provides a convenient way of producing groups of items from which collections and choices can be made. The abstract class `collection` provides a means of specifying a group of items. The subclass `choice` provides groups of selectable items, where you may specify what initial state they are in, and what happens when the selection is changed. Subclasses of `collection` and `choice` used for producing particular kinds of grouped elements are described in the sections that follow.

All the choices described in this chapter can be given a print function via the `:print-function` keyword. This allows you to control the way in which items in the element are displayed. For example, passing the argument

'string-capitalize to :print-function would capitalize the initial letters of all the words of text that an instance of a choice displays.

Some of the examples in this chapter require the functions `test-callback` and `hello` which were introduced in Chapter 4, “Creating Common Windows”.

5.1 Choice classes

This section discusses the immediate subclasses of `choice` which can be used to build button panels. If you have a group of several buttons, you can use the appropriate `button-panel` element to specify them all as a group, rather than using `push-button` or `check-button` to specify each one separately. There are three such elements altogether: `push-button-panel`, `check-button-panel` and `radio-button-panel`. The specifics of each are discussed below.

5.1.1 Push button panels

The arrangement of a number of push buttons into one group can be done with a `push-button-panel`. Since this provides a panel of buttons which do not maintain a selection when you click on them, `push-button-panel` is a `choice` that does not allow a selection. When a button is activated it causes a `:selection-callback`, but the button does not maintain the selected state.

Here is an example of a push button panel:

```
(make-instance 'push-button-panel
               :items '(one two three four five)
               :selection-callback 'test-callback
               :print-function 'string-capitalize)

(contain *)
```

Figure 5.1 A push button panel



The layout of a button panel (for instance, whether items are listed vertically or horizontally) can be specified using the `:layout-class` keyword. This can take two values: `'column-layout` if you wish buttons to be listed vertically, and `'row-layout` if you wish them to be listed horizontally. The default value is `'row-layout`. If you define your own layout classes, you can also use these as values to `:layout-class`. Layouts, which apply to many other CAPI objects, are discussed in detail in Chapter 6, “Laying Out CAPI Panes”.


5.1.2 Radio button panels

A group of radio buttons (a group of buttons of which only one at a time can be selected) is created with the `radio-button-panel` class. Here is an example of a radio button panel:

```
(setq radio (make-instance 'radio-button-panel
                          :items (list 1 2 3 4 5)
                          :selection-callback 'test-callback))

(contain radio)
```

Figure 5.2 A radio button panel



5.1.3 Check button panels

A group of check buttons can be created with the `check-button-panel` class. Any number of check buttons can be selected.

Here is an example of a check button panel:

```
(contain
  (make-instance
    'check-button-panel
    :items '("Red" "Green" "Blue")))
```

Figure 5.3 A check button panel



5.1.4 Mnemonics in button panels

On Windows and GTK+ you can specify the mnemonics (underlined letters) in a button panel with the `:mnemonics` initarg, for example:

```
(contain
  (make-instance 'push-button-panel
    :items '(one two three many)
    :mnemonics '(#\O #\T #\E :none)
    :print-function 'string-capitalize))
```

Notice that the value `:none` removes the mnemonic.

5.2 List panels

Lists of selectable items can be created with the `list-panel` class. Here is a simple example of a list panel:

```
(setq list
  (make-instance 'list-panel
    :items '(one two three four)
    :visible-min-height '(character 2)
    :print-function 'string-capitalize))
```

```
(contain list)
```

Figure 5.4 A list panel



Notice how the items in the list panel are passed as symbols, and a *print-function* is specified which controls how those items are displayed on the screen.

Any item on the list can be selected by clicking on it with the mouse.

By default, list panels are single selection — that is, only one item in the list may be selected at once. You can use the `:interaction` keyword to change this:

```
(make-instance 'list-panel
               :items (list "One" "Two" "Three" "Four")
               :interaction :multiple-selection)

(contain *)
```

You can add callbacks to any items in the list using the `:selection-callback` keyword.

```
(make-instance 'list-panel
               :items (list "One" "Two" "Three" "Four")
               :selection-callback 'test-callback)

(contain *)
```

5.2.1 List interaction

If you select different items in the list, only the last item you select remains highlighted. The way in which the items in a list panel interact upon selection can be controlled with the `:interaction` keyword.

The list produced in the example above is known as a single-selection list because only one item at a time may be selected. List panels are `:single-selection` by default.

There are also multiple-selection and extended-selection lists available. The possible interactions for list panels are:

- `:single-selection` — only one item may be selected
- `:multiple-selection` — more than one item may be selected
- `:extended-selection` — see Section 5.2.2

To get a particular interaction, supply one of the values above to the `:interaction` keyword, like this:

```
(contain
  (make-instance
    'list-panel
    :items '("Red" "Green" "Blue")
    :interaction :multiple-selection))
```

Note that `:no-selection` is not a supported choice for list panels. To display a list of items with no selection possible you should use a `display-pane`.

5.2.2 Extended selection

Application users often want to make single *and* multiple selections from a list. Some of the time they want a new selection to deselect the previous one, so that only one selection remains — just like a `:single-selection` panel. On other occasions, they want new selections to be added to the previous ones — just like a `:multiple-selection` panel.

The `:extended-selection` interaction combines these two interactions. Here is an extended-selection list panel:

```
(contain
  (make-instance
    'list-panel
    :items '("Item" "Thing" "Object")
    :interaction :extended-selection))
```

Before continuing, here are the definitions of a few terms. The action you perform to select a single item is called the *selection gesture*. The action performed to select additional items is called the *extension gesture*. There are two extension gestures. To add a single item to the selection, the extension gesture is a click of the left button while holding down the `control` key. For selecting a range of items, it is a click of the left button whilst holding down the `shift` key.

5.2.3 Deselection, retraction, and actions

As well as selecting items, users often want to deselect them. Items in multiple-selection and extended-selection lists may be deselected.

In a multiple-selection list, deselection is done by clicking on the selected item again with either of the selection or extension gestures.

In an extended-selection list, deselection is done by performing the extension gesture upon the selected item. (If this was done using the selection gesture, the list would behave as a single-selection list and all other selections would be lost.)

Just like a selection, a deselection — or *retraction* — can have a callback associated with it.

For a multiple-selection list panel, there may be the following callbacks:

- `:selection-callback` — called when a selection is made
- `:retract-callback` — called when a selection is retracted

Consider the following example. The function `set-title` changes the title of the interface to the value of the argument passed to it. By using this as the callback to the `check-button-panel`, the title of the interface is set to the current selection. The *retract-callback* function displays a message dialog with the name of the button retracted.


1. Display the example window:

```
(defun set-title (data interface)
  (setf (interface-title interface)
        (format nil "~A" (string-capitalize data))))

(make-instance 'check-button-panel
  :items '(one two three four five)
  :print-function 'string-capitalize
  :selection-callback 'set-title
  :retract-callback 'test-callback)


(contain *)
```

Figure 5.5 The example check button panel before the callback.



2. Try selecting one of the check buttons. The window title will change:

Figure 5.6 The example check button panel after the callback.



3. Now de-select the button. Notice that the *retract-callback* is called.

For an extended-selection list pane, there may be the following callbacks:

- `:selection-callback` — called when a selection is made
- `:retract-callback` — called when a selection is retracted
- `:extend-callback` — called when a selection is extended

Also available in extended-selection and single-selection lists is the action callback. This is called when you double-click on an item.

- `:action-callback` — called when a double-click occurs

5.2.4 Selections

List panels — all choices, in fact — can have selections, and you can set them from within Lisp. This is useful for providing default settings in a choice, or when a user selection has an effect on other settings than just the one they made.

The selection is represented as a vector of offsets into the list of the choice's items, unless it is a single-selection choice, in which case it is just represented as an offset.

The initial selection is controlled with the initarg `:selection`. The accessor `choice-selection` is provided.

5.2.5 Filters

You can add a filter to a `list-panel` by passing the `:filter` initarg.

List panel filters are used in the LispWorks IDE, for example in the Inspector tool.


5.3 Graph panes

Another kind of choice is the `graph-pane`. This is a special pane that can draw graphs, whose nodes and edges can be selected, and for which callbacks can be specified, as usual.

Here is a simple example of a graph pane. It draws a small rooted tree:

```
(contain
  (make-instance
    'graph-pane
    :roots '(1)
    :children-function
    #'(lambda (x)
        (when (< x 8)
          (list (* 2 x) (1+ (* 2 x)))))))
```

Figure 5.7 A graph pane



The graph pane is supplied with a `:children-function` which it uses to calculate the children of the root node, and from those children it continues to calculate more children until the termination condition is reached. For more details of this, see the *LispWorks CAPI Reference Manual*.

`graph-pane` provides a gesture which expands or collapses a node, depending on its current state. Click on the circle alongside the node to expand or collapse it.

You can associate selection, retraction, extension, and action callbacks with any or all elements of a graph. Here is a simple graph pane that has an action callback on its nodes.

First we need a pane which will display the callback messages. Executing the following form to create this pane:

```
(defvar *the-collector*
  (contain (make-instance 'collector-pane)))
```

Then, define the following four callback functions:

```
(defun test-action-callback (&rest args)
  (format (collector-pane-stream
          *the-collector*) "Action"))

(defun test-selection-callback (&rest args)
  (format (collector-pane-stream *the-collector*)
          "Selection"))

(defun test-extend-callback (&rest args)
  (format (collector-pane-stream *the-collector*)
          "Extend"))

(defun test-retract-callback (&rest args)
  (format (collector-pane-stream *the-collector*)
          "Retract"))
```

Now create an extended selection graph pane which uses each of these callbacks, the callback used depending on the action taken:

```
(contain
  (make-instance
    'graph-pane
    :interaction :extended-selection
    :roots '(1)
    :children-function
    #'(lambda (x)
        (when (< x 8)
          (list (* 2 x) (1+ (* 2 x))))))
    :action-callback 'test-action-callback
    :selection-callback 'test-selection-callback
    :extend-callback 'test-extend-callback
    :retract-callback 'test-retract-callback))
```

The selection callback function is called whenever any node in the graph is selected.

The extension callback function is called when the selection is extended by middle clicking on another node (thus selecting it too).

The `retract` callback function is called whenever an already selected node is deselected.

The `action` callback function is called whenever an action is performed on a node (that is, whenever it gets a double-click, or `Return` is pressed while the node is selected).

5.4 Option panes

Option panes, created with the `option-pane` class, display the current selection from a single-selection list. When you click on the option pane, the list appears and you can make another selection from it. Once the selection is made, it is displayed in the option pane.

The appearance of the `option-pane` list varies between platforms: a drop-down list box on Microsoft Windows; a combo box on GTK+ or Motif, and a popup list on Cocoa.

Here is an example option pane, which shows the choice of one of five numbers. The initial selection is controlled with `:selected-item`.

```
(contain
 (make-instance
  'option-pane
  :items '(1 2 3 4 5)
  :selected-item 3
  :title "One of Five:"))
```

Figure 5.8 An option pane



5.4.1 Option panes with images

You can add images to option pane items. Supply the `:image-function` initial argument when creating the `option-pane`, as illustrated in

`examples/capi/elements/option-pane-with-images.lisp`

5.5 Text input choice

A `text-input-choice` class is provided which allows arbitrary text input augmented with a choice like an `option-pane`.

See `examples/capi/elements/text-input-choice.lisp`.

5.6 Menu components

Menus (covered in Chapter 8) can have components that are also choices. These components are groups of items that have an interaction upon selection just like other choices. The `:interaction` keyword is used to associate radio or check buttons with the group — with the values `:single-selection` and `:multiple-selection` respectively. By default, a menu component has an interaction of `:no-selection`.

See “Grouping menu items together” on page 72 for more details.

5.7 General properties of choices

The behaviors you have seen so far are mostly general properties of choices rather than being specific to a particular choice. These general properties are summarized below.

5.7.1 Interaction

All choices have an interaction style, controlled by the `:interaction` initarg. The `radio-button-panel` and `check-button-panel` are simply `button-panels` with their interactions set appropriately. The possible values for *interaction* are listed below.

`:single-selection`

Only one item may be selected at a time: selecting an item deselects any other selected item.

`:multiple-selection`

A multiple selection choice allows the user to select as many items as she wants. A selected item may be deselected by clicking on it again.

`:extended-selection`

An extended selection choice is a combination of the previous two: only one item may be selected, but the selection may be extended to more than one item.

`:no-selection`

Forces no interaction. Note that this option is not available for list panels. To display a list of items with no selection you should use a display pane instead.

Specifying an interaction style that is invalid for a particular choice causes a compilation error.

The accessor `choice-interaction` is provided for accessing the *interaction* of a choice.

5.7.2 Selections

All choices have a selection. This is a state representing the items currently selected. The selection is represented as a vector of offsets into the list of the choice's items, unless it is a single-selection choice, in which case it is just represented as an offset.

The initial selection is controlled with the initarg `:selection`. The accessor `choice-selection` is provided.

Generally, it is easier to refer to the selection in terms of the items selected, rather than by offsets, so the CAPI provides the notion of a *selected item* and the *selected items*. The first of these is the selected item in a single-selection choice. The second is a list of the selected items in any choice.

The accessors `choice-selected-item` and `choice-selected-items` provide access to these conceptual slots, and you can also supply the values at `make-instance` time via the initargs `:selected-item` and `:selected-items`.

5.7.3 Callbacks

All choices can have callbacks associated with them. These callbacks are activated when the application user makes a selection, and different sorts of gesture can have different sorts of callback associated with them.

The following callbacks are available: `:selection-callback`, `:retract-callback` (called when a deselection is made), `:extend-callback`, and `:action-callback` (called when a double-click occurs). What makes one choice different from another is that they permit different combinations of these callbacks. This is a consequence of the differing interactions. For example, you cannot have an `:extend-callback` in a radio button panel, because you cannot extend selection in one.

Callbacks pass data to the function they call. There are default arguments for each type of callback. Using the `:callback-type` keyword allows you to change these defaults. Example values of *callback-type* are `:interface` (which causes the interface to be passed as an argument to the callback function), `:data` (the value of the selected data is passed), `:element` (the element containing the callback is passed) and `:none` (no arguments are passed). Also there is a variety of composite `:callback-type` values, such as `:data-interface` (which causes two arguments, the data and the interface, to be passed). See the `callbacks` entry in the *LispWorks CAPI Reference Manual* for a complete description of `:callback-type` values.

The following example uses a push button and a callback function to display the arguments it receives.

```
(defun show-callback-args (arg1 arg2)
  (display-message "The arguments were ~S and ~S" arg1 arg2))

(setq example-button
  (make-instance 'push-button
    :text "Push Me"
    :callback 'show-callback-args
    :data "Here is some data"
    :callback-type :data-interface))

(contain example-button)
```

Try changing the `:callback-type` to other values.

If you do not use the `:callback-type` argument and you do not know what the default is, you can define your callback function with lambda list (`&rest args`) to account for all the arguments that might be passed.

Specifying a callback that is invalid for a particular choice causes a compile-time error.

6

Laying Out CAPI Panes

So far, you have seen how you can create a variety of different window elements using the CAPI. Up to now, though, you have only created interfaces which contain one of these elements. The CAPI provides a series of layout elements which allow you to combine several elements in a single window. This chapter provides an introduction to the different types of layout available and the ways in which each can be used.

Layouts are created just like any other CAPI element, by using `make-instance`. Each layout must contain a description of the CAPI elements it contains, given as a list to the `:description` keyword.

A layout is used to group any instances of `simple-pane` and its subclasses (for instance all the elements you met in the last chapter), and `pinboard` object and its subclasses (discussed in Chapter 11, “Creating Your Own Panes”). Once again, you should make sure you have defined the `test-callback` function before attempting any of the examples in this chapter. Its definition is repeated here for convenience.

```
(defun test-callback (data interface)
  (display-message "Data ~S in interface ~S"
                  data interface))
```


6.1 Organizing panes in columns and rows

You will frequently need to organize a number of different elements in rows and columns. The `column-layout` and `row-layout` elements are provided to make this easy.

The following is a simple example showing the use of `column-layout`.

```
(contain (make-instance 'column-layout
  :description (list
    (make-instance 'text-input-pane)
    (make-instance 'list-panel
      :items '(1 2 3 4 5))))))
```

Figure 6.1 An example of using `column-layout`



1. Define the following elements:

```
(setq button1 (make-instance 'push-button
  :data "Button 1"
  :callback 'test-callback))
```

```
(setq button2 (make-instance 'push-button
  :data "Button 2"
  :callback 'test-callback))
```

```
(setq editor (make-instance 'editor-pane
                           :text "An editor pane"))

(setq message (make-instance 'display-pane
                             :text "A display pane"))


(setq text (make-instance 'text-input-pane
                          :title "Text: "
                          :title-position :left
                          :callback 'test-callback))
```

These will be used in the examples throughout the rest of this chapter.

To arrange any number of elements in a column, create a layout using `column-layout`, listing the elements you wish to use. For instance, to display `title`, followed by `text` and `button1`, enter the following into a Listener:

```
(contain (make-instance 'column-layout
                        :description
                        (list text button1)))
```

Figure 6.2 A number of elements displayed in a column



To arrange the same elements in a row, simply replace `column-layout` in the example above with `row-layout`. If you run this example, close the column layout window first: each CAPI element can only be on the screen once at any time.

Layouts can be given horizontal and vertical scroll bars, if desired; the keywords `:horizontal-scroll` and `:vertical-scroll` can be set to `t` or `nil`, as necessary.

When creating panes which can be resized (for instance, list panels, editor panes and so on) you can specify the size of each pane relative to the others by

listing the proportions of each. This can be done via either the `:y-ratios` keyword (for column layouts) or the `:x-ratios` keyword (for row layouts).

```
(contain (make-instance 'column-layout
  :description (list
    (make-instance 'display-pane)
    (make-instance 'editor-pane)
    (make-instance 'listener-pane))
  :y-ratios '(1 5 3)))
```

You may need to resize this window in order to see the size of each pane.

Note that the heights of the three panes are in the proportions specified. The `:x-ratios` initarg will adjust the width of panes in a row layout in a similar way.

It is also possible to specify that some panes are fixed at their minimum size whilst others in the same row or column adjust proportionately when the interface is resized:

```
(contain
  (make-instance
    'column-layout
    :description
    (list
      (make-instance 'editor-pane
        :text "Resizable"
        :visible-min-height '(:character 1))
      (make-instance 'editor-pane
        :text "Fixed"
        :visible-min-height '(:character 1))
      (make-instance 'editor-pane
        :text
        (format nil "Resizable~%Resizable~%Resizable")
        :visible-min-height '(:character 3)))
    :y-ratios '(1 nil 3)
  ))
```

To arrange panes in your row or column layout with constant gaps between them, use the `:gap` initarg:

```
(contain
  (make-instance
    'column-layout
    :description (list
      (make-instance 'output-pane
        :background :red)
      (make-instance 'output-pane
        :background :white)
      (make-instance 'output-pane
        :background :blue))
    :gap 20
    :title "Try resizing this window vertically"
    :background :grey))
```

To create resizable spaces between panes in your row or column layout, use the special value `nil` in the layout *description*:

```
(contain (make-instance 'column-layout
  :description (list
    (make-instance 'output-pane
      :background :red)
    nil
    (make-instance 'output-pane
      :background :white)
    nil
    (make-instance 'output-pane
      :background :blue))
  :y-ratios '(1 1 4 1 1)
  :title "Try resizing this window vertically"
  :background :grey))
```

6.2 Other types of layout

Row and column layouts are the most basic type of layout class available in the CAPI, and will be sufficient for many things you want to do. A variety of other layouts are available as well, as described in this section.

6.2.1 Grid layouts

Whereas row and column layouts only allow you to position a pane horizontally *or* vertically (depending on which class you use), grid layouts let you specify both, thus allowing you to create a complete grid of different CAPI panes.

`grid-layout` supports a title column, as illustrated in

```
examples/capi/layouts/titles-in-grid.lisp
```

and it supports cells spanning multiple columns or rows, as illustrated in

```
examples/capi/layouts/extend.lisp
```

6.2.2 Simple layouts

Simple layouts control the layout of only one pane. Where possible, the pane is resized to fit the layout. Simple layouts are sometimes useful when you need to encapsulate a pane.

6.2.3 Pinboard layouts

Pinboard layouts allow you to position a pane anywhere within a window, by specifying the *x* and *y* integer coordinates of the pane precisely. They are a means of letting you achieve any effect which you cannot create using the other available layouts, although their use can be correspondingly more complex. They are discussed in more detail in Chapter 11, “Creating Your Own Panes”.


6.3 Combining different layouts

You will not always want to arrange all your elements in a single row or column. You can include other layouts in the list of elements used in any layout, thus enabling you to specify precisely how panes in a window should be arranged.

For instance, suppose you want to arrange the elements in your window as shown in Figure 6.3. The two buttons are shown on the right, with the text

input pane and a message on the left. Immediately below this is the editor pane.

Figure 6.3 A sample layout




The layout in Figure 6.3 can be achieved by creating two row layouts: one containing the display pane and a button, and one containing the text input pane and the other button, and then creating a column layout which uses these two row layouts and the editor.

```
(setq row1 (make-instance 'row-layout
                          :description (list message button1)))

(setq row2 (make-instance 'row-layout
                          :description (list text button2)))
```

```
(contain (make-instance 'column-layout
                        :description
                        (list row1 row2 editor)))
```

Figure 6.4 An instantiation of the sample layout



As you can see, creating a variety of different layouts is simple. This means that it is easy to experiment with different layouts, allowing you to concentrate on the interface design, rather than its code.

However, remember that each instance of a CAPI element must not be used in more than one place at the same time.

6.4 Constraining the size of layouts

The size of a layout (often referred to as its *geometry*) is calculated automatically on the basis of the size of each of its children. The algorithm used takes account of *hints* provided by the children, and from the description of the layout itself. Hints are specified via the panes' *initargs* when they are created. The various pane classes have useful default values for these *initargs*.

6.4.1 Default Constraints

If you do not specify any hints, the CAPI calculates the on-screen geometry based on its default constraints. With this geometry the various elements are displayed with adequate space in the window.

This is designed to work regardless of variable factors such as the user's configuration, for example specifying large font sizes. It is often wrong to constrain CAPI elements to fixed pixel sizes, as these constraints may lead to poorer layouts in some configurations.

6.4.2 Width and Height Constraints

In the CAPI, there are three kinds of constraint: external, visible and internal. The following hints are recognized by all layouts:

External constraints control the size that the pane takes up in its parent:

- `:external-min-width` — the minimum width of the child in its parent
- `:external-max-width` — the maximum width of the child in its parent
- `:external-min-height` — the minimum height of the child in its parent
- `:external-max-height` — the maximum height of the child in its parent

Visible constraints control the size of the part of the pane that you can see:

- `:visible-min-width` — the minimum visible width of the child.
- `:visible-max-width` — the maximum visible width of the child.
- `:visible-min-height` — the minimum visible height of the child.
- `:visible-max-height` — the maximum visible height of the child.

Internal constraints control the size of region used to display the contents of the pane:

- `:internal-min-width` — the minimum width of the display region.
- `:internal-max-width` — the maximum width of the display region.
- `:internal-min-height` — the minimum height of the display region.
- `:internal-max-height` — the maximum height of the display region.


Initargs `:min-width`, `:max-width`, `:min-height` and `:max-height` are deprecated. They are synonyms for the visible constraints `:visible-min-width` and so on.

Each external size is the same as the visible size plus the borders.

For a non-scrolling pane, the internal constraints are the same as the visible constraints.

For a scrolling pane, the internal constraints control the size of region over which you can scroll and the visible constraints control the size of the viewport. Usually the internal constraints are computed by the widget. Here is an illustration of the external, internal and visible sizes in a scrolling pane. $ABCD$ is the external size, $abcd$ is the visible size, and $ABCD$ is the internal size:

Figure 6.5 External, visible and internal sizes:



6.4.3 Constraint Formats

Hints can take arguments in a number of formats, which are described in full in the *LispWorks CAPI Reference Manual*. When given a number, this should be an integer and the layout is constrained to that number of pixels. A constraint

can also be specified in terms of character widths or heights, as shown in the next section.

6.4.3.1 Character constraints

In “Combining different layouts” on page 52, you created a window with five panes, by combining row and column layouts. Now consider changing the definition of the editor pane so that it is required to have a minimum size. This would be a sensible change to make, because editor panes need to be large enough to work with comfortably.


```
(setq editor2
  (make-instance 'editor-pane
    :text "An editor pane with minimum size"
    :visible-min-width '(:character 30)
    :visible-min-height '(:character 10)))
```

Now display a window similar to the last example, but with the `editor2` editor pane. Note that it is only the description of the top-level column layout which differs. Before entering the following into the listener, you should close all the windows created in this chapter in order to free up the instances of `button1`, `button2` and so forth.

```
(contain (make-instance 'column-layout
  :description
  (list row1 row2 editor2)))
```

You will not be able to resize the window any smaller than this:

Figure 6.6 The result of resizing the sample layout



6.4.3.2 Changing the constraints

If you need to alter the constraints on an existing element, use the function `set-hint-table`. See how the interface above resizes after this call:

```
(apply-in-pane-process editor2
 'set-hint-table editor2 '(:visible-min-width (:character 100)))
```

6.4.3.3 String constraints

To make a pane that is wide enough to accommodate a given string, use the `:visible-min-width` hint with a `(:string string)` constraint.

In this example we also supply `:visible-max-width t`, which fixes the maximum visible width to be the same as the minimum visible width. Hence the pane is wide enough, but no wider:

```
(defvar *text* "Exactly this wide")

(capi:contain
  (make-instance 'capi:text-input-pane
    :text *text*
    :visible-min-width `(:string ,*text*)
    :visible-max-width t
    :font (gp:make-font-description
      :size (+ 6 (random 30))))))
```

Note that the width constraint works regardless of the font used.

6.5 Advanced pane layouts

Until now you have used layouts for CAPI elements in which the constituents were displayed in fixed positions set out by the CAPI. In this chapter we will be looking at a number of ways in which users can select the layout and display of CAPI elements in an interface once an instance of the interface has been displayed.

The portable techniques are the use of dividers, switchable layouts and tab layouts. On Microsoft Windows, there is also Multiple-Document Interface (MDI).

Throughout this section we will be using three predefined panes, which you should define before proceeding.

```
(setq red-pane (make-instance 'output-pane
  :background :red))

(setq green-pane (make-instance 'output-pane
  :background :green))

(setq blue-pane (make-instance 'output-pane
  :background :blue))
```

6.5.1 Switchable layouts

A switchable layout allows you to place CAPI objects on top of one another and determine which object is displayed on top through Lisp code, possibly linked to a button or menu option through a callback. Switchable layouts are set up using a `switchable-layout` element in a `make-instance`. As with the other layouts, such as `column-layout` and `row-layout`, the elements to be organized are given as a list to the `:description` keyword. Here is an example:

```
(setq switching-panes (make-instance
                      'switchable-layout
                      :description (list red-pane green-pane)))

(contain switching-panes)
```

Note that the default pane to be displayed is the red pane, which was the first pane in the description list. The two panes can now be switched between using `switchable-layout-visible-child`:

```
(apply-in-pane-process
 switching-panes #'(setf switchable-layout-visible-child)
 green-pane switching-panes)

(apply-in-pane-process
 switching-panes #'(setf switchable-layout-visible-child)
 red-pane switching-panes)
```


6.5.2 Tab layouts

In its simplest mode, a tab layout is similar to a switchable layout, except that each pane is provided with a labelled tab, like the tabs on filing cabinet folders or address books. If the tab is clicked on by the user, the pane it is attached to is pulled to the front. Don't forget to close the switchable layout window created in the last example before displaying this:

```
(make-instance 'tab-layout
              :items (list (list "one" red-pane)
                           (list "two" green-pane)
                           (list "three" blue-pane))
              :print-function 'car
              :visible-child-function 'second)
```

```
(contain *)
```

Figure 6.7 A tab layout



The example needs the `:print-function` to be `car`, or else the tabs will be labelled with the object numbers of the panes as well as the title provided in the list.

However, a tab layout can also be used in a non-switchable manner, with each tab responding with a callback to alter the appearance of only one pane. In this mode the `:description` keyword is used to describe the main layout of the tab pane. In the following example the tabs alter the choice of starting node for one graph pane, by using a callback to the `graph-pane-roots` accessor:

```

(defun tab-graph (items)
  (let* ((gp (make-instance 'graph-pane))
        (tl (make-instance 'tab-layout
                           :description (list gp)
                           :items items
                           :visible-child-function nil
                           :key-function nil
                           :print-function (lambda (x) (format nil "~R" x))
                           :callback-type :data
                           :selection-callback #'(lambda (data)
                                                  (setf (graph-pane-roots gp)
                                                      (list data))))))
    (contain tl)))

(tab-graph '(1 2 4 5 7))

```

6.5.3 Dividers and separators

If you need adjacent panes in a row or column to have a narrow user-movable horizontal divider between them, supply the special value `:divider` in the *description*. The divider allows the user to resize one pane into the space of the other. To see this in the column layout below, grab the divider between the two panes and then drag it vertically to resize both panes:

```

(contain (make-instance 'column-layout
                      :description (list green-pane
                                         :divider red-pane)))

```

The arrow keys can also be used to move the divider.

To include a narrow non-movable visible element between adjacent panes, supply the special value `:separator` in the *description*.

If you also specify ratios, the ratio for each occurrence of either of these special values should be `nil` to specify that the narrow element is fixed at its minimum size:

```
(contain (make-instance 'column-layout
  :description (list
    (make-instance 'output-pane
      :background :red)
    :divider
    (make-instance 'output-pane
      :background :white)
    :separator
    (make-instance 'output-pane
      :background :blue))
  :y-ratios '(1 nil 4 nil 1)
  :title "You can drag the divider, but not the separator"
  :background :grey))
```

Dividers and separators can also be placed between panes in a `row-layout` or even combinations of row and column layouts.

6.5.4 Multiple-Document Interface (MDI)

In LispWorks for Windows, the CAPI supports MDI through the class `document-frame`. See the entry for `document-frame` in the *LispWorks CAPI Reference Manual*.

MDI is not supported on other platforms.

7

Modifying CAPI Windows

An interface or its children can be altered in many ways. This chapter describes APIs for the most common of these.

Note: By default, each CAPI interface runs in its process. It is important to understand that an on-screen interface and its elements must be accessed only in the process of that interface. In most circumstances the user alters the interface by a callback inside the interface, which will automatically happen in the correct process. However, calls from other processes (including other CAPI interfaces) should use `execute-with-interface`, `execute-with-interface-if-alive` or `apply-in-pane-process`. See the *LispWorks CAPI Reference Manual* for details of these functions.

7.1 Initialization

If necessary you can run code after your interface's windows have been created but before they are displayed on screen. Do this by supplying `create-callback` for your interface. For example:

```

(defun make-text (self createdp)
  (multiple-value-bind (s m h dd mm yy)
    (decode-universal-time (get-universal-time))
    (format nil "Window ~S ~:[displayed~;created~] at
~2,'0D:~2,'0D:~2,'0D"
      self createdp h m s)))

(let ((editor (make-instance 'editor-pane
                            :text "Not created yet.")))
  (contain editor
    :interface-args
    (list :create-callback
          #'(lambda (self)
              (setf (editor-pane-text editor)
                    (make-text self t))))))

```

Also, if you need to run code at display time you may define a `:before` or `:after` method on the generic function `interface-display`. Your method will run just before or just after your interface is displayed on screen.

7.2 Resizing

Programmatic resizing can be done using the function `set-top-level-interface-geometry`. For example, to double the width of an interface about its center:

```
(setf interface (contain (make-instance 'interface)))
```

Use the mouse or window manager-specific gesture to resize the interface, then evaluate:

```

(multiple-value-bind (x y w h)
  (top-level-interface-geometry interface)
  (execute-with-interface interface
    'set-top-level-interface-geometry
    interface
    :x (round (- x (* 0.5 w)))
    :y y
    :width (* 2 w)
    :height h))

```

All resize operations are subject to the constraints. The constraints can be altered programmatically as described in “Changing the constraints” on page 58.

7.3 Scrolling

Programmatic scrolling is implemented with the generic function `scroll`. This example shows vertical scrolling in a `list-panel`:

```
(setf list-panel
  (contain
    (make-instance 'list-panel
      :items (loop for i below 100 collect i)
      :vertical-scroll t)))

(apply-in-pane-process
  list-panel 'scroll list-panel :vertical :move 50)
```

Elsewhere this manual shows how an `editor-pane` can be scrolled using editor commands.

7.3.1 Automatic scrolling

Automatic scrolling of the parent to show the focus pane can be specified by using `scroll-if-not-visible-p`.

7.4 Swapping panes and layouts

The class `switchable-layout` is useful when your interface has several panes of which exactly one should be visible at any time. The class `tab-layout` provides similar functionality in a Window-system specific way. See “Advanced pane layouts” on page 59

To change to another layout, use `(setf pane-layout)`:

```
(setf layout
  (capi:contain
    (make-instance 'row-layout
      :description
      (list (make-instance 'title-pane :text "One")
            (make-instance 'title-pane :text "Two")))
      :visible-min-height 100))

(apply-in-pane-process
 layout #'(setf pane-layout)
 (make-instance 'column-layout
  :description
  (list (make-instance 'title-pane :text "Three")
        (make-instance 'title-pane :text "Four")))
 (element-interface layout))
```

To change the panes within a layout, use `(setf layout-description)`:

```
(setf layout
  (capi:contain
    (make-instance 'row-layout
      :description
      (list (make-instance 'title-pane :text "One")
            (make-instance 'title-pane :text "Two")))
      :visible-min-height 100))

(apply-in-pane-process
 layout #'(setf layout-description)
 (list (make-instance 'title-pane :text "Three")
       (make-instance 'title-pane :text "Four")
       (make-instance 'title-pane :text "Five")))
 layout)
```

Note: you must not reuse already-displayed panes in a CAPI layout.

7.5 Setting the data in a pane

Use only the documented functions such as the accessors `(setf editor-pane-text)` and `(setf collection-items)` and so on. For details, see the *LispWorks CAPI Reference Manual* entry for the particular pane class and its superclasses.

7.6 Iconifying and restoring windows

You can iconify an interface window as follows:

```
(setf (top-level-interface-display-state interface) :iconic)
```

You can also make it be hidden, maximized or restore it to normal, and you have the option to create it in one of these states initially. For details see the documentation for `top-level-interface-display-state` in the *LispWorks CAPI Reference Manual*.

7.7 Destroying

To destroy a CAPI interface, call the generic function `destroy`.

7 *Modifying CAPI Windows*

8

Creating Menus

You can create menus for an application using the `menu` class.

You should make sure you have defined the `test-callback` and `hello` functions before attempting any of the examples in this chapter. Their definitions are repeated here for convenience.

```
(defun test-callback (data interface)
  (display-message "Data ~S in interface ~S"
                  data interface))

(defun hello (data interface)
  (declare (ignore data interface))
  (display-message "Hello World"))
```

8.1 Creating a menu

A menu can be created in much the same way as any of the CAPI classes you have already met.

1. Enter the following into a Listener:

```
(make-instance 'menu
  :title "Foo"
  :items '("One" "Two" "Three" "Four")
  :callback 'test-callback)
```

```
(make-instance 'interface
  :menu-bar-items (list *))

(display *)
```

This creates a CAPI interface with a menu, **Foo**, which contains four items. Choosing any of these items displays its arguments. Each item has the callback specified by the `:callback` keyword.

A submenu can be created simply by specifying a menu as one of the items of the top-level menu.

2. Enter the following into a Listener:

```
(make-instance 'menu
  :title "Bar"
  :items '("One" "Two" "Three" "Four")
  :callback 'test-callback)

(make-instance 'menu
  :title "Baz"
  :items (list 1 2 * 4 5)
  :callback 'test-callback)

(contain *)
```

This creates an interface which has a menu, called **Baz**, which itself contains five items. The third item is another menu, **Bar**, which contains four items. Once again, selecting any item returns its arguments.

Menus can be nested as deeply as required using this method.

8.2 Grouping menu items together

The `menu-component` class lets you group related items together in a menu. This allows similar menu items to share properties, such as callbacks, and to be visually separated from other items in the menus. Menu components are actually choices.


Here is a simple example of a menu component. This creates a menu called **Items**, which has four items. **Menu 1** and **Menu 2** are ordinary menu items, but **Item 1** and **Item 2** are created from a menu component, and are therefore grouped together in the menu.

```
(setq component (make-instance 'menu-component
                              :items '("item 1" "item2")
                              :print-function 'string-capitalize
                              :callback 'test-callback))

(contain (make-instance 'menu
                       :title "Items"
                       :items
                       (list "menu 1" component "menu 2")
                       :print-function 'string-capitalize
                       :callback 'hello)

         :width 150
         :height 0)
```

Figure 8.1 A menu



Menu components allow you to specify, via the `:interaction` keyword, selectable menu items — either as multiple-selection or single-selection items. This is like having radio buttons or check boxes as items in a menu, and is a popular technique among many GUI-based applications.

The following example shows you how to include a panel of radio buttons in a menu.

```
(setq radio (make-instance 'menu-component
                          :interaction :single-selection
                          :items '("This" "That")
                          :callback 'hello))

(setq commands (make-instance 'menu
                             :title "Commands"
                             :items
                             (list "Command 1" radio "Command 2")
                             :callback 'test-callback))

(contain commands)
```

Figure 8.2 Radio buttons included in a menu



The menu items **This** and **That** are radio buttons, only one of which may be selected at a time. The other menu items are just ordinary commands, as you saw in the previous examples. Note that the CAPI automatically groups the items which are parts of a menu component so that they are separated from other items in the menu.

This example also illustrates the use of more than one callback in a menu, which of course is the usual case when you are developing real applications. Choosing either of the radio buttons displays one message on the screen, and choosing either **Command1** or **Command2** returns the arguments of the callback.

Checked menu items can be created by specifying `:multiple-selection` to the `:interaction` keyword, as illustrated below.

```
(setq letters (make-instance 'menu-component
                            :interaction :multiple-selection
                            :items (list "Alpha" "Beta")))

(contain (make-instance 'menu
                       :title "Greek"
                       :items (list letters)
                       :callback 'test-callback))
```

Figure 8.3 An example of checked menu items



Note how the items in the menu component inherit the callback given to the parent, eliminating the need to specify a separate callback for each item or component in the menu.

Within a menu or component, you can specify alternatives for a main menu item that are invoked by modifier keys. See “Alternative menu items” on page 79 for more information.

8.3 Creating individual menu items

The `menu-item` class lets you create individual menu items. These items can be passed to menu-components or menus via the `:items` keyword. Using this class, you can assign different callbacks to different menu items.


```
(setq test (make-instance 'menu-item
                        :title "Test"
                        :callback 'test-callback))
```

```
(setq hello (make-instance 'menu-item
                          :title "Hello"
                          :callback 'hello))

(setq group (make-instance 'menu-component
                          :items (list test hello)))

(contain group)
```

Figure 8.4 Individual menu items



Remember that each instance of a menu item must not be used in more than one place at a time.

8.4 The CAPI menu hierarchy

The combination of menu items, menu components and menus can create a hierarchical structure as shown schematically in Figure 8.5 and graphically in Figure 8.6. This menu has five elements, one of which is itself a menu (with three menu items) and the remainder are menu components and menu items. Items in a menu inherit values from their parent, allowing similar elements to share relevant properties whenever possible.

```

(defun menu-item-name (data)
  (format nil "Menu Item ~D" data))

(defun submenu-item-name (data)
  (format nil "Submenu Item ~D" data))

(contain
 (make-instance
  'menu
  :items
  (list
   (make-instance 'menu-component
                  :items '(1 2)
                  :print-function 'menu-item-name)
   (make-instance 'menu-component
                  :items
                  (list 3
                      (make-instance
                       'menu
                       :title "Submenu"
                       :items '(1 2 3)
                       :print-function
                        'submenu-item-name))
                  :print-function 'menu-item-name)
   (make-instance 'menu-item
                  :data 42))
  :print-function 'menu-item-name))

```

Figure 8.5 A schematic example of a menu hierarchy





Figure 8.6 An example of a menu hierarchy



8.5 Mnemonics in menus

On Microsoft Windows and GTK+ you can control the mnemonics in menu titles and menu items using the initargs `:mnemonic`, `:mnemonic-title` (and if necessary `:mnemonic-escape`).

This example illustrates the various ways you can specify the mnemonics in a menu:

```

(contain
  (make-instance
    'menu
    :mnemonic-title "M&nemonics"
    :items
    (list
      (make-instance 'menu-item
        :data "Menu Item 1"
        :mnemonic #\1)
      (make-instance 'menu-item
        :data "Menu Item 2"
        :mnemonic 10)
      (make-instance 'menu-item
        :mnemonic-title "Menu Item &3")
      (make-instance 'menu-item
        :mnemonic-title "Menu Item !4"
        :mnemonic-escape #\!)
      (make-instance 'menu-item
        :data "Menu Item 5"
        :mnemonic :default)
      (make-instance 'menu-item
        :data "Menu Item 6"
        :mnemonic :none))))

```

8.6 Alternative menu items

Menus can include "alternative" items, which are invoked if some modifiers are held while selecting the "main" item. The modifiers are defined by the `:accelerator` initarg of the item, which also allows the item to be invoked by a keyboard accelerator key if specified. On Cocoa, the title and accelerator of the alternative item appear when the appropriate modifier(s) are pressed.

A menu item becomes an alternative to an immediately previous item when it is made with initarg `:alternative t`. Each alternative item must have the same parent as its previous item. That is, they are within the same menu and menu component, as described in "Grouping menu items together" on page 72. More than one alternative item can be supplied for a given main item by putting them consecutively in the menu. The main item is the item preceding the first alternative item.

The main item and its alternative items forms a group of items. The accelerators of all items in the group must consist of the same key, but with different modifiers. If there is no need for an accelerator key, the main item

should not have no accelerator and the alternative items should have accelerators with `Null` as the key, for example `"Shift-Null"`.

When the menu is displayed, only one item from the group will be shown. On Windows, GTK+ and Motif the main item is always displayed. Cocoa displays the item with the least number of modifiers initially, so to get a consistent cross-platform behavior, the main item should have the least number of modifiers. On Cocoa, pressing modifier keys that match alternative items changes the title and accelerators displayed for the item.

When the user selects an item with the modifiers pressed, the appropriate alternative item is selected.

To make a `menu-item` an alternative item, pass the `initarg :alternative t` and a suitable value for the `initarg :accelerator`.

There is an example illustrating alternative menu items in

```
examples/capi/elements/accelerators.lisp
```

Note: Accelerators of alternative items do not work on Motif.

8.7 Disabling menu items


A function can be specified via the `:enabled-function` `initarg`, that determines whether or not the menu, menu item, or menu component is enabled. By default, a menu object is always enabled.

Consider the following example:

```
(defvar *on* nil)

(contain
 (make-instance 'menu
  :items
  (list
   (make-instance
    'menu-item
    :title "Foo"
    :enabled-function
    #'(lambda (menu) *on*))
   (make-instance
    'menu-item
    :title "Bar"))))
```

Figure 8.7 A menu with a disabled menu item



Changing the value of `*on*` between `t` and `nil` in the Listener, using `setq`, results in the menu item changing between the enabled and disabled states.

8.7.1 Dialogs and disabled menu items

By default, items in the menu bar menus and sub-menus are disabled while a dialog is on the screen on top of the active window. You can override this by passing a suitable value for the `menu-item` initarg `:enabled-function-for-dialog`.

8.8 Menus with images

You can add images to menu items. Supply the `:image-function` initarg when creating the `menu`, as illustrated in

```
examples/capi/elements/menu-with-images.lisp
```

Note: on some platforms support for images in menus is limited to menu items without text and/or images without transparency. If `pane-supports-menus-with-images` returns true, then images are fully supported in menus.

8.9 Popup menus for panes

The CAPI tries to display a popup menu for a pane when the `:post-menu` gesture is entered by the user (mouse-right-click or `shift+F10` on Microsoft Windows, GTK+ or Motif, control-click on Cocoa). See below for the special case of `output-pane`.

It first tries to get a menu for the pane. There are two mechanisms by which it can get a menu: which is tried depends on the value of `pane-menu`.

1. If the pane's initial `pane-menu` is not `:default` in the call to `make-instance`, then its value is used. If the value is a function or a fbound symbol, it is called with four arguments: the pane, data (this is the selected object if there is a selection), x, y. It should return a menu. If it is not a function or a fbound symbol, it should be a menu, which is used directly. The `:pane-menu` mechanism is useful when the menu needs to be dependent on the location of the mouse inside the pane, or when each pane requires a unique menu. In other cases, the other mechanism is more useful.
2. If `pane-menu` is `:default` (this the default value), CAPI calls the generic function `make-pane-popup-menu` with two arguments: the pane and its interface. The result should be a menu.

If the chosen mechanism does not produce a menu, the CAPI does not do anything in response to `:post-menu`.

The system definition of `make-pane-popup-menu` calls `pane-popup-menu-items` with the pane and the interface, and if this returns non-nil list, it calls `make-menu-for-pane` to make the menu. You can define `make-pane-popup-menu` methods that specialize on your pane or interface classes, but in most cases it is more useful to add methods to `pane-popup-menu-items`. `make-menu-for-pane` is used to generate the menu, and it makes the menu such that by default all setup callbacks are done on the pane itself, rather than on the interface. `make-pane-popup-menu` is useful when

the application needs a menu with the same items as the items on the popup menu, typically to add it to the menu bar.

In `output-pane`, you control the input behavior using the *input-model*. By default, the system assigns `:post-menu` and `:keyboard-post-menu` (`Shift+F10`) to a callback that raises a menu as described above, but your code can override this in the *input-model*.

9

Defining Interface Classes

So far we have looked at various components for building interfaces. The CAPI provides all these and more, but instead of continuing with our exploration of the various classes provided, let us see how what we have learned so far can be combined into a single, non-trivial interface class.

9.1 The `define-interface` macro

The macro `define-interface` is used to define subclasses of `interface`, the superclass of all CAPI interface classes.

It is an extension to `defclass`, which provides the functionality of that macro as well as the specification of the panes, layouts, and menus from which an interface is composed. It takes the same arguments as `defclass`, and supports the additional options `:panes`, `:layouts`, `:menus`, and `:menu-bar`.

If you specify `:panes` but no `:layouts`, then on creating your interface the CAPI will create a `column-layout` and arrange the panes in it in the order they are defined. For real applications you will need some control over how the panes are laid out, and this is supplied via the `:layouts` option.

Each component of the interface is named in the code, and a slot of that name is added to the class created. When an instance of the class is made, each component is created automatically and placed in its slot.

To access a pane, layout or menu in an instance of your interface class you can define an accessor, like `viewer-pane` in the example below, or simply use `with-slots`.

When defining a component, you can use other components within the definition simply by giving its name. You can refer to the interface itself by the special name `capi:interface`.

9.2 An example interface

Here is a simple example of interface definition done with `define-interface`:

```
(define-interface demo ()
  ()
  (:panes
   (page-up push-button
    :text "Page Up")
   (page-down push-button
    :text "Page Down")
   (open-file push-button
    :text "Open File"))
  (:layouts
   (row-of-buttons row-layout
    '(page-up page-down open-file)))
  (:default-initargs :title "Demo"))
```

An instance of this interface can be displayed as follows:

```
(make-instance 'demo)


(display *)
```

At the moment the buttons do nothing, but they will eventually do the following:

- **Open File** will bring up a file prompter and allow you to select a file-name from a directory. Later on, we will add an editor pane to display the chosen file's contents.
- **Page Down** will scroll downwards so that you can view the lower parts of the file that cannot be seen initially.

- **Page Up** will scroll upwards so that you can return to parts of the file seen before.

Figure 9.1 A demonstration of a CAPI interface



Later on, we will specify callbacks for these buttons to provide this functionality.

The `(:default-initargs :title "Demo")` part at the end is necessary to give the interface a title. If no title is given, the default name is “Untitled CAPI Interface”.

Note: the `define-interface` form could be generated by the Interface Builder tool in the LispWorks IDE. See the *LispWorks IDE User Guide* for details. As the interface becomes more complex, you will find it more convenient to edit the definition by hand.

9.2.1 How the example works

Examine the `define-interface` call to see how this interface was built. The first part of the call to define interface is shown below:

```
(define-interface demo ()
  ())
```

This part of the macro is identical to `defclass` — you provide:

- The name of the interface class being defined
- The superclasses of the interface (defaulting to `interface`)
- The slot descriptions

The interesting part of the `define-interface` call occurs after these `defclass`-like preliminaries. The remainder of a `define-interface` call lists all elements

that define the interface's appearance. Here is the `:panes` part of the definition:

```
(:panes
  (page-up push-button
    :text "Page Up")
  (page-down push-button
    :text "Page Down")
  (open-file push-button
    :text "Open File"))
```

Two arguments — the name and the class — are required to produce a pane. You can supply slot values as you would for any pane.

Here is the `:layouts` part of the definition:

```
(:layouts
  (row-of-buttons row-layout
    '(page-up page-down open-file)))
```

Three arguments — the name, the class, and any child layouts — are required to produce a layout. Notice how the children of the layout are specified by using their component names.

The interface information given so far is a series of specifications for panes and layouts. It could also specify menus and a menu bar. In this case, three buttons are defined. The layout chosen is a row layout, which displays the three buttons side by side at the top of the pane.

9.3 Adapting the example

The `:panes` and `:layouts` keywords can take a number of panes and layouts, each specified one after the other. By listing several panes, menus, and so on, complicated interfaces can be constructed quickly.

To see how simply this is done, let us add an editor pane to our interface. We need this to display the text contained in the file chosen with the **Open File** button.

The editor pane needs a layout. It could be added to the `row-layout` already built, or another layout could be made for it. Then, the two layouts would have to be put inside a third to contain them (see Chapter 6, *Laying Out CAPI Panes*).

The first thing to do is add the editor pane to the panes description. The old panes description read:

```
(:panes
  (page-up push-button
    :text "Page Up")
  (page-down push-button
    :text "Page Down")
  (open-file push-button
    :text "Open File"))
```

The new one includes an editor pane named `viewer`.

```
(:panes
  (page-up push-button
    :text "Page Up")
  (page-down push-button
    :text "Page Down")
  (open-file push-button
    :text "Open File")
  (viewer editor-pane
    :title "File:"
    :text "No file selected."
    :visible-min-height '(:character 8)
    :reader viewer-pane))
```

This specifies the editor pane, with a stipulation that it must be at least 8 characters high. This allows you to see a worthwhile amount of the file being viewed in the pane.

Note the use of `:reader`, which defines a reader method for the interface which returns the editor pane. Similarly, you can also specify writers or accessors. If you omit accessor methods, it is still possible to access panes and other elements in an interface instance using `with-slots`.

The interface also needs a layout for the editor pane in the layouts section. The old layouts description read:

```
(:layouts
  (row-of-buttons row-layout
    '(page-up page-down open-file)))
```

The new one reads:

```

(:layouts
 (main-layout column-layout
   '(row-of-buttons row-with-editor-pane))
 (row-of-buttons row-layout
   '(page-up page-down open-file))
 (row-with-editor-pane row-layout
   '(viewer)))

```

This creates another `row-layout` for the new pane and then encapsulates the two row layouts into a third `column-layout` called `main-layout`. This is used as the default layout, specified by setting the `:layout` initarg to `main-layout` in the `:default-initargs` section. If there is no default layout specified, `define-interface` uses the first one listed.

By putting the layout of buttons and the layout with the editor pane in a column layout, their relative position has been controlled: the buttons appear in a row above the editor pane.

The code for the new interface is now as follows:

```


(define-interface demo ()
  ()
  (:panes
   (page-up push-button
    :text "Page Up")
   (page-down push-button
    :text "Page Down")
   (open-file push-button
    :text "Open File")
   (viewer editor-pane
    :title "File:"
    :text "No file selected."
    :visible-min-height '(:character 8)
    :reader viewer-pane))
  (:layouts
   (main-layout column-layout
    '(row-of-buttons row-with-editor-pane))
   (row-of-buttons row-layout
    '(page-up page-down open-file))
   (row-with-editor-pane row-layout
    '(viewer)))
  (:default-initargs :title "Demo"))

```

Displaying an instance of the interface by entering the line of code below produces the window in Figure 9.2:

```
(display (make-instance 'demo))
```

Figure 9.2 A CAPI interface with editor pane



9.3.1 Adding menus

To add menus to your interface you must first specify the menus themselves, and then a menu bar of which they will be a part.

Let us add some menus that duplicate the proposed functionality for the buttons. We will add:

- A **File** menu with a **Open** option, to do the same thing as **Open File**
- A **Page** menu with **Page Up** and **Page Down** options, to do the same things as the buttons with those names

The extra code needed in the `define-interface` call is this:

```
(:menus
  (file-menu "File"
             ("Open"))
  (page-menu "Page"
             ("Page Up" "Page Down")))
(:menu-bar file-menu page-menu)
```

Menu definitions give a slot name for the menu, followed by the title of the menu, a list of menu item descriptions, and then, optionally, a list of keyword arguments for the menu.

In this instance the menu item descriptions are just strings naming each item, but you may wish to supply initialization arguments for an item — in which case you would enclose the name and those arguments in a list.

The menu bar definition simply names all the menus that will be on the bar, in the order that they will appear. By default, of course, the environment may add menus of its own to an interface — for example the **Works** menu in the LispWorks IDE.


The code for the new interface is:

```

(define-interface demo ()
  ()
  (:panes
   (page-up push-button
    :text "Page Up")
   (page-down push-button
    :text "Page Down")
   (open-file push-button
    :text "Open File")
   (viewer editor-pane
    :title "File:"
    :text "No file selected."
    :visible-min-height '(:character 8)
    :reader viewer-pane))
  (:layouts
   (main-layout column-layout
    '(row-of-buttons row-with-editor-pane))
   (row-of-buttons row-layout
    '(page-up page-down open-file))
   (row-with-editor-pane row-layout
    '(viewer)))
  (:menus
   (file-menu "File"
    ("Open"))
   (page-menu "Page"
    ("Page Up" "Page Down")))
  (:menu-bar file-menu page-menu)
  (:default-initargs :title "Demo"))

```

Figure 9.3 A CAPI interface with menu items



The menus contain the items specified — try it out to be sure.

9.4 Connecting an interface to an application

Having defined an interface in this way, you can connect it up to your program using callbacks, as described in earlier chapters. Here we define some functions to perform the operations we required for the buttons and menus, and then hook them up to the buttons and menus as callbacks.

The functions to perform the page scrolling operations are given below:

```
(defun scroll-up (data interface)
  (call-editor (viewer-pane interface)
    "Scroll Window Up"))

(defun scroll-down (data interface)
  (call-editor (viewer-pane interface)
    "Scroll Window Down"))
```

The functions use the generic function `call-editor` which calls an editor command (given as a string) on an instance of an `editor-pane`. The editor commands **Scroll Window Up** and **Scroll Window Down** perform the necessary operations for **Page Up** and **Page Down** respectively.

The function to perform the file-opening operation is given below:

```
(defun file-choice (data interface)
  (let ((file (prompt-for-file "Select A File:")))
    (when file
      (setf (titled-object-title (viewer-pane interface))
            (format nil "File: ~S" file))
      (setf (editor-pane-text (viewer-pane interface))
            (with-open-file (stream file)
              (let ((buffer
                    (make-array 1024
                               :element-type
                               (stream-element-type stream)
                               :adjustable t
                               :fill-pointer 0)))
                (do ((char (read-char stream nil nil)
                        (read-char stream nil nil)))
                    ((null char))
                  (vector-push-extend char buffer))
                (subseq buffer 0))))))))))
```

This function prompts for a filename and then displays the file in the editor pane.

The function first produces a file prompter through which a file may be selected. Then, the selected file name is shown in the title of the editor pane (using `titled-object-title`). Finally, the file name is used to get the contents of the file and display them in the editor pane (using `editor-pane-text`).

The correct callback information for the buttons is specified as shown below:

```

(:panes
  (page-up push-button
    :text "Page Up"
    :selection-callback 'scroll-up)
  (page-down push-button
    :text "Page Down"
    :selection-callback 'scroll-down)
  (open-file push-button
    :text "Open File"
    :selection-callback 'file-choice)
  (viewer editor-pane
    :title "File:"
    :text "No file selected."
    :visible-min-height '(:character 8)
    :reader viewer-pane))

```

All the buttons and menu items operate on the editor pane `viewer`. A reader is set up to allow access to it.

The correct callback information for the menus is specified as shown below:

```

(:menus
  (file-menu "File"
    ("Open"))
    :selection-callback 'file-choice)
  (page-menu "Page"
    ("Page Up"
     :selection-callback 'scroll-up)
    ("Page Down"
     :selection-callback 'scroll-down)))

```

In this case, each item in the menu has a different callback. The complete code for the interface is listed below — try it out.

```

(define-interface demo ()
  ()
  (:panes
   (page-up push-button
    :text "Page Up"
    :selection-callback 'scroll-up)
   (page-down push-button
    :text "Page Down"
    :selection-callback 'scroll-down)
   (open-file push-button
    :text "Open File"
    :selection-callback 'file-choice)
   (viewer editor-pane
    :title "File:"
    :text "No file selected."
    :visible-min-height '(:character 8)
    :reader viewer-pane))
  (:layouts
   (main-layout column-layout
    '(row-of-buttons row-with-editor-pane))
   (row-of-buttons row-layout
    '(page-up page-down open-file))
   (row-with-editor-pane row-layout
    '(viewer)))
  (:menus
   (file-menu "File"
    ("Open"))
    :selection-callback 'file-choice)
   (page-menu "Page"
    ("Page Up"
     :selection-callback 'scroll-up)
    ("Page Down"
     :selection-callback 'scroll-down))))
  (:menu-bar file-menu page-menu)
  (:default-initargs :title "Demo"))

```

9.5 Controlling the interface title

You can add dynamic control of window titles using the functions illustrated in the section.

Firstly we add a counter to the title of new `demo` windows:

```
(defvar *demo-title-counter* 0)

(defmethod capi:interface-extend-title ((self demo) title)
  (format nil "~A - ~D"
          (call-next-method)
          (incf *demo-title-counter*)))

(capi:display (make-instance 'demo))
```

Then we specify common prefix for all interface window titles:

```
(capi:set-default-interface-prefix-suffix
 :prefix "My " :suffix nil )

(capi:display (make-instance 'demo))
```

10

Prompting for Input

A dialog is a window that receives some input from the user and returns it to the application. For instance, if the application wants to know where to save a file, it could prompt the user with a file dialog. Dialogs can also be cancelled, meaning that the application should cancel the current operation.

In order to let you know whether or not the dialog was cancelled, CAPI dialog functions always return two values. The first value is the return value itself, and the second value is `τ` if the dialog returned normally and `nil` if the dialog was cancelled.

On Cocoa you can control whether a CAPI dialog is application-modal or window-modal. In the latter case the user can work with the application's other windows while the dialog is on screen.

The CAPI provides both a large set of predefined dialogs and the means to create your own. This chapter takes you through some example uses of the predefined dialogs, and then shows you how to create custom built dialogs.

The last section briefly describes a way to get input for completions via a special non-modal window.

10.1 Some simple dialogs

The simplest form of dialog is a message dialog. The function `display-message` behaves very much like `format`.

```
(display-message "Hello world")
```

Figure 10.1 A message dialog



```
(display-message  
  "This function is ~S"  
  'display-message)
```


Figure 10.2 A second message dialog



Another simple dialog asks the user a question and returns `t` or `nil` depending on whether the user has chosen yes or no. This function is `confirm-yes-or-no`.

```
(confirm-yes-or-no  
  "Do you own a pet?")
```

Figure 10.3 A message dialog prompting for confirmation



For more control over such a dialog, use the function `prompt-for-confirmation`. See the *LispWorks CAPI Reference Manual* for details.

10.2 Prompting for values


The CAPI provides a number of different dialogs for accepting values from the user, ranging from accepting strings to accepting whole Lisp forms to be evaluated.

10.2.1 Prompting for strings

The simplest of the CAPI prompting dialogs is `prompt-for-string` which returns the string you enter into the dialog.

```
(prompt-for-string  
 "Enter a string:")
```

Figure 10.4 A dialog prompting for a string



An initial value can be placed in the dialog by specifying the keyword argument `:initial-value`.

10.2.2 Prompting for numbers

The CAPI also provides a number of more specific dialogs that allow you to enter other types of data. For example, to enter an integer, use the function `prompt-for-integer`. Only integers are accepted as valid input for this function.

```
(prompt-for-integer  
 "Enter an integer:")
```

There are a number of extra options which allow you to specify more strictly which integers are acceptable. Firstly, there are two arguments `:min` and `:max` which specify the minimum and maximum acceptable integers.

```
(prompt-for-integer  
 "Enter an integer in the inclusive range [10,20]:"  
 :min 10 :max 20)
```

If this does not provide enough flexibility you can specify a function that validates the result with the keyword argument `:ok-check`. This function is passed the current value and must return non-nil if it is a valid result.

```
(prompt-for-integer
 "Enter an odd integer:"
 :ok-check 'oddp)
```


Try also the function `prompt-for-number`.

10.2.3 Prompting for an item in a list

If you would like the user to select an item from a list of items, the function `prompt-with-list` should handle the majority of cases. The simplest form just passes a list to the function and expects a single item to be returned.

```
(prompt-with-list  
'(:red :yellow :blue)  
"Select a color:")
```

Figure 10.5 A dialog prompting for a selection from a list



You can also specify the interaction style that you would like for your dialog, which can be any of the interactions accepted by a choice. The specification of the interaction style to this choice is made using the keyword argument `:interaction:`

```
(prompt-with-list
 '(:red :yellow :blue)
 "Select a color:"
 :interaction :multiple-selection)
```

By default, the dialog is created using a list-panel to display the items, but the keyword argument `:choice-class` can be specified with any choice pane. Thus, for instance, you can present a list of buttons.

```
(prompt-with-list
 '(:red :yellow :blue)
 "Select a color:"
 :interaction :multiple-selection
 :choice-class 'button-panel)
```

Figure 10.6 Selection from a button panel



Finally, as with any of the prompting functions, you can specify additional arguments to the pane that has been created in the dialog. Thus to create a column of buttons instead of the default row, use:

```
(prompt-with-list
 '(:red :yellow :blue)
 "Select a color:"
 :interaction :multiple-selection
 :choice-class 'button-panel
 :pane-args
 '(:layout-class column-layout))
```

Figure 10.7 Selection from a column of buttons



10.2.4 Prompting for files


To prompt for a file, use the function `prompt-for-file`:

```
(prompt-for-file
 "Enter a file:")
```

You can also specify a starting pathname:

```
(prompt-for-file
 "Enter a filename:"
 :pathname "/tmp/")
```

Figure 10.8 Selection of a file



Try also the function `prompt-for-directory`.

10.2.5 Prompting for fonts

To obtain a `gp:font` object from the user call `prompt-for-font`.

10.2.6 Prompting for colors

To obtain a color specification from the user call `prompt-for-color`.

10.2.7 Prompting for Lisp objects

The CAPI provides a number of dialogs specifically designed for creating Lisp aware applications. The simplest is the function `prompt-for-form` which accepts an arbitrary Lisp form and optionally evaluates it.

```
(prompt-for-form
 "Enter a form to evaluate:"
 :evaluate t)

(prompt-for-form
 "Enter a form (not evaluated):"
 :evaluate nil)
```

Another useful function is `prompt-for-symbol` which prompts the user for an existing symbol. The simplest usage accepts any symbol, as follows:

```
(prompt-for-symbol
 "Enter a symbol:")
```

If you have a list of symbols from which to choose, then you can pass `prompt-for-symbol` this list with the keyword argument `:symbols`.

Finally, using `:ok-check` you can accept only certain symbols. For example, to only accept a symbol which names a class, use:

```
(prompt-for-symbol
 "Enter a class-name symbol:"
 :ok-check #'(lambda (symbol)
              (find-class symbol nil)))
```

Cocoa programmers will notice that the dialog sheet displayed by this form, like all those in this chapter so far, prevents input to other LispWorks windows while it is displayed. For information about creating dialog sheets which are not application-modal, see “Window-modal Cocoa dialogs” on page 108.

10.3 Window-modal Cocoa dialogs

By default, CAPI dialogs on Cocoa use sheets which are application-modal. This means that the application waits until the sheet is dismissed and does not allow the user to work with its other windows until then.

This section describes how to create CAPI dialogs which are window-modal on Cocoa. This is done with portable code, so Windows, GTK+ and Motif programmers may wish to code their CAPI dialogs as described in this section, which would ease a future port to the Cocoa GUI.

10.3.1 The `:continuation` argument

All CAPI dialog functions take a keyword argument *continuation*. This is a function which is called with the results of the dialog.

You do not need to construct the continuation argument yourself, but rather call the dialog function inside `with-dialog-results`.

10.3.2 A dialog which is window-modal on Cocoa

To create a dialog which is window-modal on Cocoa, call the dialog function inside the macro `with-dialog-results` as in this example:

```
(with-dialog-results (symbol okp)
  (prompt-for-symbol
   "Enter a class-name symbol:"
   :ok-check #'(lambda (symbol)
                 (find-class symbol nil)))
  (when okp
    (display-message "symbol is ~S" symbol)))
```

On Microsoft Windows, GTK+ and Motif this displays the dialog, calls `display-message` when the user clicks **OK**, and then returns. The effect is no different to what you saw in “Prompting for Lisp objects” on page 108.

On Cocoa, this creates a sheet and returns. `display-message` is called when the user clicks **OK**. The sheet is window-modal, unlike the sheet you saw in “Prompting for Lisp objects” on page 108.

For more details, see the page for `with-dialog-results` in the *LispWorks CAPI Reference Manual*.

10.4 Dialog Owners

When a dialog appears, it should be "owned" by some window. The main effect of this "ownership" is that the dialog is always in front of the owner window. When either the dialog or the owner is raised, the other follows.

All CAPI functions which display a dialog allow you to specify the owner.

10.4.1 The default owner

When a dialog is displayed and the owner is not supplied or is given as `nil`, the CAPI tries to identify the appropriate owner. In particular, in the case where a dialog pops up in a process in which a CAPI interface is displayed, by default the CAPI uses this interface as the owner window. This case covers most situations.

10.4.2 Specifying the owner

If the default is not appropriate, then the programmer needs to supply the owner. This *owner* argument can be any CAPI pane that is currently displayed, and the top level interface of the pane is used as the actual owner. A CAPI pane owner must be running in the current process (see the *process* argument to `display`). Creating cross-process ownership can lead to deadlocks.

The *owner* can also be a `screen` object, which tells the system on which screen to put the dialog, but none of the windows will be the dialog's owner.

The *owner* can be supplied by the keyword argument `:owner` in functions such as `display-dialog` and `print-dialog`. Other functions such as `prompt-for-string` and `prompt-for-file` can be supplied an owner in the `:popup-args` list as a pair `:owner owner`.

10.5 Creating your own dialogs

The CAPI provides a number of built-in dialogs which should cover the majority of most peoples needs. However, there is always the occasional need to create custom built dialogs, and the CAPI makes this very simple, using the function `display-dialog` which displays any CAPI interface as a dialog, and the functions `exit-dialog` and `abort-dialog` as the means to return from such a dialog.


10.5.1 Using `display-dialog`

Here is a very simple example that displays a **Cancel** button in a dialog, and when that button is pressed the dialog is cancelled. Note that `display-dialog`

must receive an interface, so an interface is created for the button by using the function `make-container`.

```
(display-dialog
 (make-container
  (make-instance
   'push-button
   :text "Press this button to cancel"
   :callback 'abort-dialog)
  :title "My Dialog"))
```

Figure 10.9 A cancel button



The function `abort-dialog` cancels the dialog returning the values `nil` and `nil`, which represent a return result of `nil` and the fact that the dialog was cancelled, respectively. Note also that `abort-dialog` accepts any values and just ignores them.

The next problem is to create a dialog that can return a result. Use the function `exit-dialog` which returns the value passed to it from the dialog. The example below shows a simple string prompter.

```
(display-dialog
 (make-container
  (make-instance
   'text-input-pane
   :callback-type :data
   :callback 'exit-dialog)
  :title "Enter a string:"))
```

Both of these examples are very simple, so here is a slightly more complicated one which creates a column containing both a text-input-pane and a **Cancel** button.

```
(display-dialog
  (make-container
    (list
      (make-instance
        'text-input-pane
        :callback-type :data
        :callback 'exit-dialog)
      (make-instance
        'push-button
        :text "Cancel"
        :callback 'abort-dialog))
    :title "Enter a string:"))
```

Note that this looks very similar to the dialog created by `prompt-for-string` except for the fact that it does not provide the standard **OK** button.

It would be simple to add an **OK** button in the code above, but since almost every dialog needs these standard buttons, the CAPI provides a higher level function called `popup-confirmer` that adds the standard buttons for you. Also it arranges for the **OK** and **Cancel** buttons to respond to the `Return` and `Escape` keys respectively. `popup-confirmer` is discussed in the next section.

10.5.2 Using `popup-confirmer`

The function `popup-confirmer` is a higher level function provided to add the standard buttons to user dialogs, and it is nearly always used in preference to `display-dialog`. In order to create a dialog using `popup-confirmer`, all you need to do is to supply a pane to be placed inside the dialog along with the buttons and the title. The function also expects a title, like all of the prompter functions described earlier.

```
(popup-confirmer
  (make-instance
    'text-input-pane
    :callback-type :data
    :callback 'exit-dialog)
  "Enter a string")
```

A common thing to want to do with a dialog is to get the return value from some state in the pane specified. For instance, in order to create a dialog that prompts for an integer the string entered into the `text-input-pane` would need to be converted into an integer. It is possible to do this once the dialog has returned, but `popup-confirmer` has a more convenient mechanism. The func-

tion provides a keyword argument, `:value-function`, which gets passed the pane, and this function should return the value to return from the dialog. It can also indicate that the dialog cannot return by returning a second value which is non-nil.


In order to do this conversion, `popup-confirmer` provides an alternative exit function to the usual `exit-dialog`. This is called `exit-confirmer`, and it does all of the necessary work on exiting.

You now have enough information to write a primitive version of `prompt-for-integer`.

```
(defun text-input-pane-integer (pane)
  (let* ((text
         (text-input-pane-text pane))
        (integer
         (parse-integer
          text
          :junk-allowed t)))
    (or (and (integerp integer) integer)
        (values nil t))))

(defun popup-confirmer
  (make-instance
   'text-input-pane
   :callback 'exit-confirmer)
  "Enter an integer:"
  :value-function 'text-input-pane-integer)
```

Figure 10.10 A example using `popup-confirmer`



Note that the dialog's **OK** button never becomes activated, yet pressing `Return` once you have entered a valid integer *will* return the correct value. This is because the **OK** button is not being dynamically updated on each keystroke in

the `text-input-pane` so that it activates when the `text-input-pane` contains a valid integer. The activation of the **OK** button is recalculated by the function `redisplay-interface`, and the CAPI provides a standard callback, `:redisplay-interface`, which calls this as appropriate.

Thus, to have an **OK** button that becomes activated and deactivated dynamically, you need to specify the change-callback for the `text-input-pane` to be `:redisplay-interface`.

```
(popup-confirmer
 (make-instance
  'text-input-pane
  :change-callback :redisplay-interface
  :callback 'exit-confirmer)
 "Enter an integer:"
 :value-function 'text-input-pane-integer)
```

Note that the **OK** button now changes dynamically so that it is only ever active when the text in the `text-input-pane` is a valid integer.

Note that the **Escape** key activates the **Cancel** button - this too was set up by `popup-confirmer`.

The next thing that you might want to do with your integer prompter is to make it accept only certain values. For instance, you may only want to accept negative numbers. This can be specified to `popup-confirmer` by providing a validation function with the keyword argument `:ok-check`. This function receives the potential return value (the value returned by the value function) and it must return non-nil if that value is valid. Thus to accept only negative numbers we could pass `minusp` as the `:ok-check`.

```
(popup-confirmer
 (make-instance
  'text-input-pane
  :change-callback :redisplay-interface
  :callback 'exit-confirmer)
 "Enter an integer:"
 :value-function 'text-input-pane-integer
 :ok-check 'minusp)
```

10.5.3 Modal and non-modal dialogs

By default `popup-confirmer` and `display-dialog` create modal dialog windows which prevent input to other application windows until they are dismissed by the user clicking on a button or another appropriate gesture. You can change this behavior by passing the *modal* keyword argument.

10.6 In-place completion

'In-place completion' allows the user to select from a list of possible completions displayed in a special non-modal window which does not grab the input focus. The user can continue typing her input to reduce the number of possible completions, or use certain gestures including `Up`, `Down` and `Return` which operate on the special window.

10.6.1 In-place completion in the LispWorks IDE

The LispWorks Editor uses this functionality to display the special window near the cursor position, which is called in-place completion.

Tools in the LispWorks IDE that ask for a named object (for example, the Class Browser and the Generic Function Browser) in a text input pane have in-place completion which is invoked by the `Up` and `Down` keys.

Set the **Preferences... Environment > General > Use in-place completion** option to use in-place completion.

10.6.2 Programmatic control of in-place completion

You can add this functionality to your application as described in this section.

10.6.2.1 Text input panes

A `text-input-pane` will do in-place completion if you pass either of these `initargs`:

`:file-completion` with value `t` or a pathname designator, or

`:in-place-completion-function` with value a suitable function designator

You can add a filter to the in-place window by passing the initarg `:in-place-filter`. Additionally you can control the functionality for file completion by passing `:directories-only` and `:ignore-file-suffices`. The keyword arguments `:complete-do-action` and `:gesture-callbacks` also interact with in-place completion.

The in-place completion can be invoked explicitly for a `text-input-pane` by calling `text-input-pane-in-place-complete`.

See the *LispWorks CAPI Reference Manual* for details.

10.6.2.2 Editor panes

An `editor-pane` does in-place completion when your code calls the function `editor:complete-in-place`.

10.6.2.3 Other CAPI panes

You can also implement in-place completion on arbitrary CAPI panes by calling `prompt-with-list-non-focus`.

11

Creating Your Own Panes

The CAPI provides a wide range of built-in panes, but it is still fairly common to need to create panes of your own. In order to do this, you need to specify both the input behavior of the pane (how it reacts to keyboard and mouse events) and its output behavior (how it displays itself). The class `output-pane` is provided for this purpose.


An `output-pane` is a fully functional graphics port. This allows it to use all of the graphics ports functionality to create graphics, and it also has a powerful input model which allows it to receive mouse and keyboard input.

11.1 Displaying graphics

The following is a simple example demonstrating how to create an `output-pane` and then how to draw a circle on it.

```
(setq output-pane
  (contain
    (make-instance 'output-pane)
    :best-width 300
    :best-height 300))
```


Figure 11.1 An empty output pane



Now you can draw a circle in the empty output pane by using the graphics ports function `draw-circle`. Note that the drawing function must be called in the process of the interface containing the output pane:

```
(capi:apply-in-pane-process
 output-pane 'gp:draw-circle output-pane 100 100 50)
```

Figure 11.2 An output pane containing a circle



Notice that this circle is not permanently drawn on the `output-pane`, and when the window is next redisplayed it vanishes. To prove this to yourself, force the window to be redisplayed (for example by iconifying or resizing it). At this point, you can draw the circle again yourself but it will not happen automatically.

```
(capi:apply-in-pane-process
 output-pane 'gp:draw-circle output-pane 100 100 50)
```

In order to create a permanent display, you need to provide a function to the `output-pane` that is called to redraw sections of the output-pane when they are exposed. This function is called the *display-callback*, and it is automatically called in the correct process. When the CAPI needs to redisplay a region of an `output-pane`, it calls that output pane's *display-callback* function, passing it the `output-pane` and the region in question.

For example, to create a pane that has a permanent circle drawn inside it, do the following:

```
(defun draw-a-circle (pane x y
                     width height)
  (gp:draw-circle pane 100 100 50))

(contain
 (make-instance
  'output-pane
  :display-callback 'draw-a-circle)
 :best-width 300
 :best-height 300)
```

Notice that the callback in this example ignores the region that needs redrawing and just redraws everything. This is possible because the CAPI clips the drawing to the region that needs redisplaying, and hence only the needed part of the drawing gets done. For maximum efficiency, it would be better to only draw the minimum area necessary.

The arguments `:best-width` and `:best-height` specify the initial width and height of the interface. More detail can be found in the *LispWorks CAPI Reference Manual*.

Now that we can create output panes with our own display functions, we can create a new class of window by using `defclass` as follows.

```
(defclass circle-pane (output-pane)
  ()
  (:default-initargs
   :display-callback 'draw-a-circle))

(contain
 (make-instance 'circle-pane))
```

11.2 Receiving input from the user


You now know enough to be able to create new classes of window which can display arbitrary graphics, but to be able to create interactive windows you need to be able to receive events. The CAPI supports this through the use of an input model, which is a mapping of events to the callbacks that should be run when they occur.

When the event callback is called, it gets passed the `output-pane` and the `x` and `y` integer coordinates of the mouse pointer at the time of the event. A few events also pass additional information as necessary; for example, keyboard events also pass the key that was pressed.

For example, we can create a very simple drawing pane by adding a callback to draw a point whenever the left button is dragged across the pane. This is done as follows:

```
(contain
 (make-instance
  'output-pane
  :input-model '(((motion :button-1)
                  gp:draw-point))))
```

Figure 11.3 An interactive output pane



The input model above seems quite complicated, but it is just a list of event to callback mappings, where each one of these mappings is a list containing an event specification and a callback. An event specification is also a list containing keywords specifying the type of event required.

There is an example input model in

```
examples/capi/graphics/pinboard.lisp
```

See the manual page for `output-pane` in the *LispWorks CAPI Reference Manual* for the full *input-model* syntax..

11.3 Creating graphical objects

A common feature needed by an application is to have a number of objects displayed in a window and to make events affect the object underneath the cursor. The CAPI provides the ability to create graphical objects, to place them into a window at a specified size and position, and to display them as necessary. Also a function is provided to determine which object is under any given point so that events can be dispatched correctly.

These graphical objects are called *pinboard objects*, as they can only be displayed if they are contained within a `pinboard-layout`. To define a `pinboard-object`, you define a subclass of `drawn-pinboard-object` and specify a drawing routine for it (and you can also specify constraints on the size of your object). You can then make instances of these objects and place them into layouts just as if they were ordinary panes. You can also place these objects inside layouts as long as there is a `pinboard-layout` higher up the layout hierarchy that contains the panes.

Note: `pinboard-objects` are implemented as graphics on a native window. Compare this with `simple-pane` and its subclasses, where each instance is itself a native window. A consequence of this is that `simple-panes` do not work well within a `pinboard-layout`, since they always appear above the `pinboard-objects`. For example, to put labels on a pinboard, use `item-pinboard-object` rather than `display-pane` or `title-pane`.

Here is an example of the built-in pinboard object class `item-pinboard-object` which displays its text like a `title-pane`. Note that the function `contain` always creates a `pinboard-layout` as part of the wrapper for the object to be contained, and so it is possible to test the display of `pinboard-objects` in just the same way as you can test other classes of CAPI object.

```
(contain
  ;; CONTAIN makes a pinboard-layout if needed, so we don't
  ;; need one explicitly in this example.
  ;; You will need an explicit pinboard-layout if you define
  ;; your own interface class.
  (make-instance
   'item-pinboard-object
   :text "Hello world"))
```

Figure 11.4 A pinboard object



There is another example illustrating `item-pinboard-object` in the file

```
examples/capi/graphics/pinboard-object-text-pane.lisp
```

11.3.1 Buffered drawing

Where the display of an `output-pane` is complex you may see flickering on screen on some platforms. Typically this occurs in a `pinboard-layout` with many `pinboard-objects`, or some other characteristic that makes the display complex.

The flickering can be avoided by passing the `draw-with-buffer` initarg which causes the drawing to go to an off-screen pixmap buffer. The screen is then updated from the buffer.

Note: GTK+ and Cocoa always buffer, so the `draw-with-buffer` initarg is ignored on these platforms.

11.3.2 The implementation of graph panes

One of the major uses the CAPI itself makes of `pinboard-objects` is to implement graph panes. The `graph-pane` itself is a `pinboard-layout` and it is built using `pinboard-objects` for the nodes and edges. This is because each node (and sometimes each edge) of the graph needs to react individually to the user. For instance, when an event is received by the `graph-pane`, it is told


which pinboard object was under the pointer at the time, and it can then use this information to change the selection.

Create the following `graph-pane` and notice that every node in the graph is made from an `item-pinboard-object` as described in the previous section and that each edge is made from a `line-pinboard-object`.

```
(defun node-children (node)
  (when (< node 16)
    (list (* node 2)
          (1+ (* node 2)))))

(contain
 (make-instance
  'graph-pane
  :roots '(1)
  :children-function 'node-children)
 :best-width 300 :best-height 400)
```


Figure 11.5 A graph pane with pinboard object nodes



As mentioned before, `pinboard-layouts` can just as easily display ordinary panes inside themselves, and so the `graph-pane` provides the ability to specify the class used to represent the nodes. As an example, here is a `graph-pane` with the nodes made from `push-buttons`.

```
(contain
 (make-instance
  'graph-pane
  :roots '(1)
  :children-function 'node-children
  :node-pinboard-class 'push-button)
 :best-width 300 :best-height 400)
```

Figure 11.6 A graph pane with push-button nodes



11.3.3 An example pinboard object

To create your own pinboard objects, the class `drawn-pinboard-object` is provided, which is a `pinboard-object` that accepts a *display-callback* to display itself. The following example creates a new subclass of `drawn-pinboard-object` that displays an ellipse.

```

(defun draw-ellipse-pane (gp pane
                        x y
                        width height)

  (with-geometry pane
    (let ((x-radius
          (1- (floor %width% 2)))
          (y-radius
          (1- (floor %height% 2))))
      (gp:draw-ellipse
       gp
       (1+ (+ %x% x-radius))
       (1+ (+ %y% y-radius))
       x-radius y-radius
       :filled t
       :foreground
       (if (> x-radius y-radius)
           :red
           :yellow))))))

(defclass ellipse-pane
  (drawn-pinboard-object)
  ()
  (:default-initargs
   :display-callback 'draw-ellipse-pane
   :visible-min-width 50
   :visible-min-height 50))

(contain
 (make-instance 'ellipse-pane)
 :best-width 200
 :best-height 100)

```

Figure 11.7 An ellipse-pane class




The `with-geometry` macro is used to set the size and position, or geometry, of the ellipse drawn by the `draw-ellipse-pane` function. The fill color depends on the radii of the ellipse - try resizing the window to see this. See the *Lisp-Works CAPI Reference Manual* for more details of `drawn-pinboard-object`.

Now that you have a new ellipse-pane class, you can create instances of them and place them inside layouts. For instance, the example below creates nine ellipse panes and place them in a three by three grid.

```
(contain
  (make-instance
    'grid-layout
    :description
    (loop for i below 9
      collect
        (make-instance 'ellipse-pane))
    :columns 3)
  :best-width 300
  :best-height 400)
```

Figure 11.8 Nine ellipse-pane instances in a layout



12

Graphics Ports

12.1 Introduction

Graphics Ports allow users to write source-compatible applications for different host window systems. Graphics Ports are the destinations for drawing primitives. They are implemented with a generic host-independent part and a small host-specific part.

Graphics Ports implement a set of drawing functions and a mechanism for specifying the graphics state to be used in each drawing function call. There are four categories of graphics ports: on-screen, pixmap, printer, and metafile.

On-screen ports correspond to visible windows. Pixmap ports can be used for double-buffering graphics whereby you draw to the off-screen pixmap port and then copy the contents to the screen with `pixblt`. Printer ports are used for drawing to a printer. Metafile ports are used for recording drawing operations so that they can be realized later.

The off-screen ports (pixmap, printer and metafile) are usually created transiently by a modification of an on-screen port using `with-pixmap-graphics-port`, `with-print-job` or `with-external-metafile` respectively. The on-screen port is modified so that drawing is redirected as appropriate.

See the *LispWorks CAPI Reference Manual* for full reference entries on all the Graphics Ports functions, macros, classes and types.

12.1.1 The package

All graphics port symbols are interned in and exported from the `graphics-ports` package, nicknamed `gp`.

12.1.2 The system

The graphics ports system is available in the default LispWorks image. All symbols in the `gp` package are available to you as soon as you start LispWorks.

12.1.3 Creating instances

Pixmap graphics ports are usually required only temporarily and the macro `with-pixmap-graphics-port` allocates one for you from a cache of such objects.

12.2 Features

The main features of graphics-ports are:

1. Each port has a “graphics state” which holds all the information about drawing parameters such as line thickness, fill pattern, line-end-style and so on. A graphics state object can also be created independently of any particular graphics port.
2. The graphics state contents can either be enumerated in each drawing function call, bound to values for the entirety of a set of calls, or permanently changed.
3. The graphics state includes a transform which implements generalized coordinate transformations on the port’s coordinates.
4. Off-screen ports can compute the horizontal and vertical bounds of the results of a set of drawing function calls, thus facilitating image or pixmap generation.

12.3 Graphics state

The graphics-state object associated with each port holds values for parameters such as *foreground*, *background*, *operation*, *thickness*, *scale-thickness*, *mask* and *font* which affect graphics ports drawing to that port.

The full set of parameters is described under `make-graphics-state` in the *LispWorks CAPI Reference Manual*.

12.3.1 Setting the graphics state

The graphics state values associated with a drawing function call to a graphics port are set by one of three mechanisms.

1. Enumeration in the drawing function call. For example:

```
(draw-line port 1 1 100 100
          :thickness 10
          :scale-thickness nil
          :foreground :red)
```

2. Bound using the `with-graphics-state` macro. For example:

```
(with-graphics-state (port :thickness 10
                          :scale-thickness nil
                          :foreground :red)
  (draw-line port 1 1 100 100)
  (draw-rectangle port 2 2 40 50 :filled t))
```

3. Set by the `set-graphics-state` function. For example:

```
(set-graphics-state port :thickness 10
                     :scale-thickness nil
                     :foreground :red)
```

The first two mechanisms change the graphics state temporarily. The last one changes it permanently, effectively altering the “default” state.

12.4 Graphics state transforms

Coordinate systems for windows generally have the origin (0,0) positioned at the upper left corner of the window with X positive to the right and Y positive downwards. This is the “window coordinates” system. Generalized coordinates are implemented using scaling, rotation and translation

operations such that any Cartesian coordinates can be used within a window. The Graphics Ports system uses a transform object to achieve this.

12.4.1 Generalized points

An (x, y) coordinate pair can be transformed to another coordinate system by scaling, rotation and translation. The first two can be implemented using 2×2 matrices to hold the coefficients:

If the point P is (a, b) and it is transformed to the point $Q (a', b')$

$$P \Rightarrow Q \text{ or } (a, b) \Rightarrow (a', b')$$

$$a' = pa + rb, b' = qa + sb.$$

$$Q = PM, \text{ where } M = \begin{vmatrix} p & q \\ r & s \end{vmatrix}$$

Translation can be included in this if the points P and Q are regarded as 3-vectors instead of 2-vectors, with the 3rd element being unity:

$$\begin{aligned} Q &= PM \\ &= (a \ b \ 1) \begin{vmatrix} p & q & 0 \\ r & s & 0 \\ u & v & 1 \end{vmatrix} \end{aligned}$$

The coefficients u and v specify the translation.

So, the six elements ($p, q, r, s, u,$ and v) of the 3×3 matrix contain all the transformation information. These elements are stored in a list in the graphics state slot `transform`.

Transforms can be combined by matrix multiplication to effect successions of translation, scaling and rotation operations.

Functions are provided in Graphics Ports which apply translation, scaling and rotation to a transform, combine transforms by pre- or post-multiplication, invert a transform, and so on.

12.4.2 Drawing functions

The scan-line conversions of the drawing functions are very much host-dependent. In other words, you cannot assume that, for example (`draw-point port x y`) has exactly the same effect on all machines. Some machines might put pixels down and to the right of integer coordinates ($x y$) while others may center the pixel at ($x y$).

Note: On GTK+ and Motif the *stipple*, *pattern*, *mask-x*, and *mask-y* slots of the graphics state are also used.

See also the *LispWorks CAPI Reference Manual* entries for `draw-circle` (which draws a circle) and `draw-ellipse` (which draws an ellipse) and other drawing functions.

12.5 Graphics state operation

The value in the operation slot of the graphics state determines how the foreground color is combined with the existing color for each pixel that is drawn.

The allowed values of *operation* are the values of the Common Lisp constants `bool-e-1`, `bool-e-and` and so on. These are the allowed values of the first argument to the Common Lisp function `bool-e`. See the specification of `bool-e` in the ANSI Common Lisp standard for the full list of operations.

The color combination corresponds to the logical operation defined there, as if by calling

```
(bool-e operation new-pixel screen-pixel)
```

For example, passing `:operation bool-e-andc2` provides a graphics state where graphics ports drawing functions draw with the bitwise AND of the foreground color and the complement of the existing color of each pixel.

Note: Graphics State *operation* is not supported by Cocoa/Core Graphics so this slot is ignored on Cocoa.

12.6 Pixmap graphics ports

Pixmap graphics ports are drawing destinations which exist only as pixel arrays whose contents are not directly accessible. They can be drawn to using the *draw-thing* functions and images can be loaded using `load-image`, and their contents can be copied onto other graphics ports. However this copying can be meaningless unless the conversion of colors uses the same color device on both ports. Because color devices are associated with regular graphics ports (Windows) rather than pixmap graphics ports, you have to connect a pixmap graphics port to a regular graphics port for color conversion. This is the purpose of the *owner* slot in `pixmap-graphics-port-mixin`. The conversion of colors to pixel values is done in the same way as for regular graphics ports, but the pixmap graphics port's owner is used to find a color device. You can draw to pixmap graphics ports using pre-converted colors to avoid color conversion altogether, in which case a null color owner is OK for a pixmap graphics port.

12.6.1 Relative drawing in pixmap graphics ports

Many of the drawing functions have a *relative* argument. If non-`nil`, it specifies that when drawing functions draw to the pixmap, the extremes of the pixel coordinates reached are accumulated. If the drawing strays beyond any edge of the pixmap port (into negative coordinates or beyond its width or height), then the drawing origin is shifted so that it all fits on the port. If the drawing extremes exceed the total size available, some are inevitably lost. If *relative* is `nil`, any part of the drawing which extends beyond the edges of the pixmap is lost. If *relative* is `nil` and *collect* non-`nil`, the drawing bounds are collected for later reading, but no relative shifting of the drawing is performed. The collected bounds are useful when you need to know the graphics motion a series of drawing calls causes. The *rest* args are host-dependent. They usually include a `:width` and `:height` pair.

12.7 Portable font descriptions

Portable font descriptions are designed to solve the following problems:

- Specify enough information to uniquely determine a real font.
- Query which real fonts match a partial specification.

- Allow font specification to be recorded and reused in a later run.

All the functions described below are exported from the `gp` package.

You can obtain the names of all the fonts which are available for a given pane by calling `list-all-font-names`, which returns a list of partially-specified font descriptions.

Portable font descriptions are used only for lookup of real fonts and for storing the parameters to specify when doing a font lookup operation. To draw text in a specified font using the Graphics Ports drawing functions, supply in the graphics state a font object as returned by `find-matching-fonts` and `find-best-font`.

12.7.1 Font attributes and font descriptions

Font attributes are properties of a font, which can be combined to uniquely specify a font on a given platform. There are some portable attributes which can be used on all platforms; other attributes are platform specific and will be ignored or signal errors when used on the wrong platform.

Font descriptions are externalizable objects which contain a set of font attributes. When using a font description in a font lookup operation, missing attributes are treated as wildcards (as are those with value `:wild`) and invalid attributes signal errors. The result of a font lookup contains all the attributes needed to uniquely specify a font on that platform.

The `:stock` font attribute is special: it can be used to reliably look up a system font on all platforms.

Font descriptions can be manipulated using the functions `merge-font-descriptions` and `augment-font-description`.

These are the current set of portable font attributes and their portable types:

Table 12.1 Set of portable font attributes

Attribute	Possible values	Comments
<code>:family</code>	string	Values are not portable.

Table 12.1 Set of portable font attributes

Attribute	Possible values	Comments
<code>:weight</code>	(member <code>:normal</code> <code>:bold</code>)	
<code>:slant</code>	(member <code>:roman</code> <code>:italic</code>)	
<code>:size</code>	(or (eql <code>:any</code>) (integer 0 *))	<code>:any</code> means a scalable font
<code>:stock</code>	(member <code>:system-font</code> <code>:system-fixed-font</code>)	Stock fonts are guaranteed to exist.
<code>:charset</code>	keyword	

12.7.2 Fonts

Fonts are the objects which are actually used in drawing operations. They are made by a font lookup operation on a pane, using a font description as a pattern.

Examples of font lookup operations are `find-best-font` and `find-matching-fonts`.

Once a font object is resolved you can read its properties such as height, width and average width. The functions `get-font-height`, `get-font-width` and `get-font-average-width` and so on need a pane that is on the screen. In general, you need to call these functions within a *display-callback*, or possibly a *create-callback*. See `capi:interface` for more information about these initargs.

12.8 Working with images

Graphics Ports supports drawing a wide range of image types via your code. Also, several CAPI classes support the same image types.

To draw an image with Graphics Ports, you need an `image` object which is associated with an instance of `capi:output-pane` (or a subclass of this). You can create a `image` object from:

- A file of recognized image type
- A registered image identifier (see “Registering images”)

- An `external-image` object
- An on-screen port

Draw the image to the pane by calling `draw-image`. Certain images ("Plain Images") can be manipulated via the Image Access API. The image should be freed by calling `free-image` when you are done with it.

`capi:image-pinboard-object`, `capi:button`, `capi:list-view`, `capi:tree-view` and `capi:toolbar` all support images. There is also limited support for images in `capi:menu`. These classes handle the drawing and freeing for you.

12.8.1 Supported image types

Graphics Ports supports Bitmaps and compressed bitmaps (BMP and DIB files). It uses Operating System-specific libraries to support a range of other image types as follows:

Table 12.2 Image library and supported image types

OS	Library	Supported Image Types
Microsoft Windows	<code>gdiplus.dll</code>	GIF, JPEG, PNG, TIFF, EMF
Mac OS X	Cocoa	GIF, JPEG, TIFF, PICT and many others. Also EPS, PDF
GTK+	<code>gdk-pixbuf</code>	GIF, JPEG, PNG, TIFF and many others.
X/Motif	<code>imlib</code>	GIF, JPEG, PNG, TIFF, XPM, PGM, PPM

Functions which load images from a file attempt to identify the image type from the file type. Files of type `bmp` and `dib` are loaded by LispWorks code, files of other types are loaded by the Operating System-specific library.

Note: On Motif LispWorks uses the freeware `imlib` library which seems to be in all Linux distributions. However it is not in some UNIX systems, so you may need to install it.

Note: On Microsoft Windows LispWorks additionally supports Windows Icon files - see `load-icon-image` in the *LispWorks CAPI Reference Manual*.

Note: LispWorks 4.3 and previous versions supported only Bitmap images.

12.8.2 External images

An External Image is an intermediate object. It is a representation of a graphic but is not associated with a port and cannot be used directly for drawing.

An object of type `external-image` is created by reading an image from a file, or by externalizing an `image` object, or by copying an existing `external-image`. Or, if you have the image bitmap data, you can create one directly as in the `examples/capi/buttons/button.lisp` example.

The `external-image` contains the bitmap data, potentially compressed. You can copy `external-image` objects, or write them to file, or compress the data.

You cannot query the size of the image in an `external-image` object directly. To get the dimensions without actually drawing it on screen see “Pixmap graphics ports” on page 136.

12.8.2.1 Transparency

An External Image representing an image with a color map can specify a transparent color. When converted and drawn, this color is drawn using the background color of the port.

You can specify the transparent color by

```
(gp:read-external-image file :transparent-color-index 42)
```

or by

```
(setf
 (gp:external-image-transparent-color-index external-image)
 42)
```

You can use an image tool such as the Gimp (www.gimp.org) to figure out the transparent color index.

Note: `transparent-color-index` works only for images with a color map - those with 256 colors or less.

12.8.2.2 Converting an external image

Convert an `external-image` to an object of type `image` ready for drawing to a port in several ways as described in “Loading images”. Such conversions are cached but you can remove the caches by `clear-external-image-conversions`.

You can also convert an `image` to an `external-image` by calling `externalize-image`.

12.8.3 Registering images

One way to load an `image` is via a registered image identifier.

To establish a registered image identifier call `register-image-translation`:

```
(gp:register-image-translation
 'info-image
 (gp:read-external-image "info.bmp"
 :transparent-color-index 7))
```

You can then do:

```
(gp:load-image port 'info-image)
```

to obtain the `image` object.

12.8.4 Loading images

To create an `image` object suitable for drawing on an given pane, use one of `convert-external-image`, `read-and-convert-external-image`, `load-image`, `make-image-from-port`, `make-sub-image` or (on Microsoft Windows) `load-icon-image`.

Images need to be freed after use. When the pane that an image was created for is destroyed, the image is freed automatically. However if you want to remove the image before the pane is destroyed, you must make an explicit call `free-image`. If the image is not freed, then a memory leak will occur.

Another way to create an `image` object is to supply a registered image identifier in a CAPI class that supports images. For example you can specify an `image` in a `capl:image-pinboard-object`. Then, an `image` object is created implicitly

when the pinboard object is displayed and freed implicitly when the pinboard object is destroyed.

In all cases, the functions that create the `image` object require the pane to be already created. So if you are displaying the image when first displaying your window, take care to create the `image` object late enough, for example in the `:create-callback` of the interface or in the first `:display-callback` of the pane.

12.8.5 Querying image dimensions

To obtain the pixel dimensions of an image, load the image using `load-image` and then use the readers `image-width` and `image-height`. The first argument to `load-image` must be a pane in a displayed interface.

To query the dimensions before displaying anything you can create and "display" an interface made with the `:display-state :hidden` initarg. Call `load-image` with this hidden interface and your `external-image` object, and then use the readers `image-width` and `image-height`.

12.8.6 Drawing images

The function to draw an image is `draw-image`.

This must be called in the same process as the pane. Use `capi:apply-in-pane-process` or `capi:execute-with-interface` if necessary.

12.8.7 Image access

You can read and write pixel values in an `image` via an Image Access object, but only if the image is a Plain Image. You can ensure you have a Plain Image by using the result of

```
(load-image pane image :force-plain t)
```

To read and/or write pixel values, follow these steps:

1. Start with a Graphics Port (for example a `capi:output-pane`) and an `image` object associated with it, which is a Plain Image. See above for how to create an image object.
2. Construct an Image Access object by calling `make-image-access`.

3. To read pixels from the image, first call `image-access-transfer-from-image` on the image access object. This notionally transfers all the pixel data from the window system into the access object. It might do nothing if the window system allows fast access to the pixel data directly. Then call `image-access-pixel` with the coordinates of each pixel. The pixel values are like those returned from `color:convert-color` and can be converted to RGB using `color:unconvert-color` if required.
4. To write pixels to the image, you must have already called `image-access-transfer-from-image`. Then call `(setf image-access-pixel)` with the coordinates of each pixel to write, and then call `image-access-transfer-to-image` on the Image Access object. This notionally transfers all the pixel data back to the window system from the access object. It might do nothing if the window system allows fast access to the pixel data directly.
5. Free the image access object by calling `free-image-access` on it.

There is an example that demonstrates the uses of Image Access objects in:

```
examples/capi/graphics/image-access.lisp
```

This further example demonstrates the uses of Image Access objects with colors that have an alpha component:

```
examples/capi/graphics/image-access-alpha.lisp
```

12.8.8 Creating external images from Graphics Ports operations

To create an `external-image` object from an on-screen window, use `with-pixmap-graphics-port` as in this example:

```
(defun record-picture (output-pane)
  (gp:with-pixmap-graphics-port
    (port output-pane
      400 400
      :clear t
      :background :red)
    (gp:draw-rectangle port 0 0 200 200
      :filled t
      :foreground :blue)
    (let ((image (gp:make-image-from-port port)))
      (gp:externalize-image port image))))
```

Here *output-pane* must be a displayed instance of `capi:output-pane` (or a subclass). The code does not affect the displayed pane.

If you do not already display a suitable output pane, you can create an invisible one like this:

```
(defun record-picture-1 ()
  (let* ((p1 (make-instance 'capi:pinboard-layout))
        (win (capi:display
              (make-instance 'capi:interface
                            :display-state :hidden
                            :layout p1))))
    (progn (record-picture p1)
          (capi:destroy win))))
```

Note: There is no reason to create and destroy the invisible interface each time a new picture is recorded, so for efficiency you could cache the interface object and use it repeatedly.

13

The Color System

13.1 Introduction

The LispWorks Color System allows applications to use keyword symbols as aliases for colors in Graphics Ports drawing functions. They can also be used for backgrounds and foregrounds of windows and CAPI objects.

For example, the call

```
(gp:draw-line my-port x1 y1 x2 y2
  :foreground :navyblue)
```

uses the keyword symbol `:navyblue` for the color of the line.

Colors are looked up in a color database. The LispWorks image is delivered with a large color database already loaded (approximately 660 entries.) The color database contains color-specs which define the colors in terms of a standard color model. When the drawing function is executed, the color-spec is converted into a colormap index (or “pixel” value).

The LispWorks Color System has facilities for:

- Defining new color aliases in one of several color models
- Loading the color database from a file of color descriptions
- Converting color specifications between color models

- Defining new color models

It is accessible from the `color` package, and all symbols mentioned in this chapter are assumed to be external to this package unless otherwise stated. You can qualify them all explicitly in your code, for example `color:apropos-color-names`.

However it is more convenient to create a package which has the `color` package on its package-use-list:

```
(defpackage "MY-PACKAGE"
  (:add-use-defaults t)
  (:use "COLOR" "CAPI")
)
```

This creates a package in which all the `color` symbols (and for convenience, `capi` as well) are accessible. To run the examples in this chapter, evaluate the form above and then:

```
(in-package "MY-PACKAGE")
```

The color-models available by default are RGB, HSV and GRAY.

13.1.1 Rendering of colors

Some colors do not render exactly as expected in some CAPI classes such as `capi:title-pane` - it depends on the palette provided by the rendering system.

However, `capi:output-pane` and its subclasses support non-standard palettes.

13.2 Reading the color database

To find out what colors are defined in the color database, use the functions `apropos-color-names`. For example:

```
(apropos-color-names "RED") =>
(:ORANGERED3 :ORANGERED1 :INDIANRED3 :INDIANRED1
 :PALEVIOLETRED :RED :INDIANRED :INDIANRED2
 :INDIANRED4 :ORANGERED :MEDIUMVIOLETRED
 :VIOLETRED :ORANGERED2 :ORANGERED4 :RED1 :RED2 :RED3
 :RED4 :PALEVIOLETRED1 :PALEVIOLETRED2 :PALEVIOLETRED3
 :PALEVIOLETRED4 :VIOLETRED3 :VIOLETRED1 :VIOLETRED2
 :VIOLETRED4)
```

For information about only aliases or only original entries, use `apropos-color-alias-names` or `apropos-color-spec-names` respectively.

To get a list of all color names in the color database, call `get-all-color-names`.

13.3 Color specs

A color spec is an object which numerically defines a color in some color-model. For example the object returned by the call:

```
(make-rgb 0.0s0 1.0s0 0.0s0) =>
#(:RGB 0.0s0 1.0s0 0.0s0)
```

defines the color green in the RGB color model. (Note that short-floats are used; this results in the most efficient color conversion process. However, any floating-point number type can be used.)

To find out what color-spec is associated with a color name, use the function `get-color-spec`. It returns the color-spec associated with a symbol. If there is no color-spec associated with *color-name*, this function returns `nil`. If *color-name* is the name of a color alias, the color alias is dereferenced until a color-spec is found.

Color-specs are made using standard functions `make-rgb`, `make-hsv` and `make-gray`. For example:

```
(make-rgb 0.0s0 1.0s0 0.0s0)
(make-hsv 1.2s0 0.5s0 0.9s0)
(make-gray 0.66667s0)
```

To create a color spec with an alpha component using the above constructors, pass an extra optional argument. For example this specifies green with 40% transparency:

```
(make-rgb 0.0s0 1.0s0 0.0s0 0.6s0)
```

Note that the alpha component is only supported on Cocoa and Windows.

The predicate `color-spec-p` can be used to test for color-spec objects. The function `color-spec-model` returns the model in which a color-spec object has been defined.

13.4 Color aliases

You can enter a color alias in the color database using the function `define-color-alias`. You can remove an entry in the color database using `delete-color-translation`.

`define-color-alias` makes an entry in the color database under a name, which should be a symbol. LispWorks by convention uses keyword symbols. The name points to either a color-spec or another color name (symbol):

```
(define-color-alias :wire-color :darkslategray)
```

Attempting to replace an existing color-spec in the color database results in an error. By default, replacement of existing aliases is allowed but there is an option to control this (see the *LispWorks CAPI Reference Manual* entry for `define-color-alias`).

`delete-color-translation` removes an entry from the color-database. Both original entries and aliases can be removed:

```
(delete-color-translation :wire-color)
```

As described in Section 13.3 on page 147, the function `get-color-spec` returns the color-spec associated with a color alias. The function `get-color-alias-translation` returns the ultimate color name for an alias:

```
(define-color-alias :lispworks-blue
  (make-rgb 0.70s0 0.90s0 0.99s0))
(define-color-alias :color-background
  :lispworks-blue)
(define-color-alias :listener-background
  :color-background)
```

```
(get-color-alias-translation :listener-background)
=> :lispworks-blue
(get-color-alias-translation :color-background)
=> :lispworks-blue
```

There is a system-defined color alias `:transparent` which is useful when specified as the *background* of a pane. It is currently supported only on Cocoa. For example:

```
(capi:popup-confirmer
 (make-instance 'capi:display-pane
               :text
               (format nil "The background of this pane~%is
transparent")
               :background :transparent)
 "")
```

13.5 Color models

Three color models are defined by default: RGB, HSV and GRAY. RGB and HSV allow specification of any color within conventional color space using three orthogonal coordinate axes, while gray restricts colors to one hue between white and black. All color models contain an optional alpha component, though this is used only on Cocoa and Windows.

Table 13.1 Color models defined by default

Model	Name	Component: Range
RGB	Red Green Blue	RED (0.0 to 1.0) GREEN (0.0 to 1.0) BLUE (0.0 to 1.0) ALPHA (0.0 to 1.0)
HSV	Hue Saturation Value	HUE (0.0 to 5.99999) SATURATION (0.0 to 1.0) VALUE (0.0 to 1.0) ALPHA (0.0 to 1.0)

Table 13.1 Color models defined by default

Model	Name	Component: Range
GRAY	Gray	GRAY (0.0 to 1.0) ALPHA (0.0 to 1.0)

The Hue value in HSV is mathematically in the open interval [0.0 6.0). All values must be specified in floating point values.

You can convert color-specs between models using the available `ensure-<model>` functions. For example:

```
(setf green (make-rgb 0.0 1.0 0.0))
=> #(:RGB 0.0 1.0 0.0)
(eq green (ensure-rgb green)) => T

(ensure-hsv green) => #(:HSV 3.0 0.0 1.0)
(eq green (ensure-hsv green)) => NIL

(ensure-rgb (ensure-hsv green)) => #(:RGB 0.0 1.0 0.0)
(eq green (ensure-rgb (ensure-hsv green))) => NIL
```

Of course, information can be lost when converting to GRAY:

```
(make-rgb 0.3 0.4 0.5) => #(:RGB 0.3 0.4 0.5)
(ensure-gray (make-rgb 0.3 0.4 0.5))
=> #(:GRAY 0.39999965)
(ensure-rgb (ensure-gray
             (make-rgb 0.3 0.4 0.5)))
=> #(:RGB 0.39999965 0.39999965 0.39999965)
```

There is also `ensure-color` which takes two color-spec arguments. It converts if necessary the first argument to the same model as the second. For example:

```
(ensure-color (make-gray 0.3) green)
=> #(:RGB 0.3 0.3 0.3)
```

`ensure-model-color` takes a model as the second argument. For example:

```
(ensure-model-color (make-gray 0.3) :hsv)
=> #(:HSV 0 1.0 0.3)
```

The function `colors=` compares two color-spec objects for color equality.

Conversion to pixel values is done by `convert-color`.

13.6 Loading the color database

You can load new color definitions into the color database using `read-color-db` and `load-color-database`.

Given a color definition file `my-colors.db` of lines like these:

```
#(:RGB 1.0s0 0.980391s0 0.980391s0)      snow
#(:RGB 0.972548s0 0.972548s0 1.0s0)      GhostWhite
```

call

```
(load-color-database (read-color-db "my-colors.db"))
```

To clear the color database use the form:

```
(setf *color-database* (make-color-db))
```

Note: You should do this before starting the LispWorks IDE (that is, before `env:start-environment` is called) or before your application's GUI starts. Be sure to load new color definitions for all the colors used in the GUI. The initial colors were obtained from the `config/colors.db` file.

You can remove a color database entry with `delete-color-translation`.

13.7 Defining new color models

Before using the definition described here, you should evaluate the form:

```
(require "color-defmodel")
```

The macro `define-color-models` can be used to define new color models for use in the color system.

The default color models are defined by the following form:

```
(define-color-models ((:rgb (red 0.0 1.0)
                           (green 0.0 1.0)
                           (blue 0.0 1.0))
 (:hsv (hue 0.0 5.99999)
        (saturation 0.0 1.0)
        (value 0.0 1.0))
 (:gray (level 0.0 1.0))))
```

For example, to define a new color model YMC and keep the existing RGB, HSV and GRAY models:

```
(define-color-models ((:rgb (red 0.0 1.0)
                        (green 0.0 1.0)
                        (blue 0.0 1.0))
                    (:hsv (hue 0.0 5.99999)
                        (saturation 0.0 1.0)
                        (value 0.0 1.0))
                    (:gray (level 0.0 1.0))
                    (:ymc (yellow 0.0 1.0)
                        (magenta 0.0 1.0)
                        (cyan 0.0 1.0))))
```

You must then define some functions to convert YMC color-specs to other color-specs. In this example, those functions are named

```
make-ymc-from-rgb
make-ymc-from-hsv
make-ymc-from-gray
```

and

```
make-rgb-from-ymc
make-hsv-from-ymc
make-gray-from-ymc
```

You can make this easier, of course, by defining the functions

```
make-ymc-from-hsv
make-ymc-from-gray
make-hsv-from-ymc
make-gray-from-ymc
```

in terms of `make-ymc-from-rgb` and `make-rgb-from-ymc`.

If you never convert between YMC and any other model, you need only define the function `make-rgb-from-ymc`.

14

Printing from the CAPI—the Hardcopy API

The CAPI hardcopy API is a mechanism for printing a Graphics Port (and hence a CAPI `output-pane`) to a printer. It is arranged in a hierarchy of concepts: printers, print jobs, pagination and outputting.

Printers correspond to the hardware accessible to the OS. Print jobs control connection to a printer and any printer-specific initialization. Pagination controls the number of pages and which output appears on which page. Outputting is the operation of drawing to a page. This is accomplished using the standard Graphics Ports drawing functions and is not discussed here.

14.1 Printers

You can obtain the current printer, or ask the user to select one, by using `current-printer`. You can ask the user about configuration by using the functions `page-setup-dialog` and `print-dialog` which display the standard Page Setup and Print dialogs.

You can pass the printer object (as returned by `current-printer` or `print-dialog`) to APIs with a *printer* argument, such as `with-print-job`, `page-setup-dialog` and `print-dialog`. The printer object itself is opaque but you can modify the configuration programmatically using `set-printer-options`.

14.1.1 Standard shortcut keys in printer dialogs

On Cocoa by default the standard shortcuts `Command+P` and `Command+Shift+P` invoke **Print...** and **Page Setup...** menu commands respectively.

In Microsoft Windows editor emulation by default the standard shortcut `Ctrl+P` invokes a **Print...** menu command.

14.2 Printer definition files

On GTK+ and Motif, CAPI uses its own printer definition files to keep information about printers. These files contain a few configuration settings, and the name of the PPD file if applicable (see below for information about PPD files). When a user saves a printer configuration, the system writes such file. Note that because the printer definition file contains the name of the PPD file, it must only be moved between machines with care: the PPD file must exist in the same path.

14.3 PPD files

To fully use the functionality of a Postscript printer on GTK+ and Motif, the system needs a Postscript Printer Description (PPD) file, which is a file in a standard format defined by Adobe. It describes the options the printer has and how to control them.

When a print dialog is presented to the user (either by an explicit call to `print-dialog`, or by printing), the system uses the PPD file to find what additional options to present, and how to communicate them to the printer.

A PPD file should be supplied by the manufacturer with the printer itself. Otherwise, it is normally possible to obtain the PPD file from the website of the manufacturer. The name of a PPD file should be *printername.ppd*.

When the user configures a new printer, the first thing the system does is to show the user all the PPD files that it can find under the `*ppd-directory*` (directly, or one level of directories below it). The application should set this variable to the appropriate directory.

If the value of `*ppd-directory*` is `nil`, the system looks at the directory obtained by evaluating `(sys:lispworks-dir "postscript/ppd")`.

If the printer does not have a PPD file, the user can still use it by selecting the default button in the print dialog. This means that the system will let the user change only the basic properties of the printer, without using its more complex features.

14.4 Print jobs

A Print job is contained within a use of the macro `with-print-job`, which handles connection to the printer and sets up a graphics port for drawing to the printer.

14.5 Handling pages—page on demand printing

In *Page on Demand Printing*, the application provides code to print an arbitrary page. The application should be prepared to print pages in any order. This is the preferred means of implementing printing. Page on Demand printing uses the `with-document-pages` macro, which iterates over all pages in the document.

14.6 Handling pages—page sequential printing

Page Sequential Printing may be used when it is inconvenient for the application to implement Page on Demand printing. In Page Sequential Printing, the application prints each page of the document in order. Page on Demand printing uses the `with-page` macro, with each invocation of the `with-page` macro contributing a new page to the document.

14.7 Printing a page

In either mode of printing, the way in which a page is printed is the same. A suitable transformation must be established between the coordinate system of the `output-pane` or `printer-port` object and the physical page being printed. The page is then drawn using normal Graphics Ports operations.

14.7.1 Establishing a page transform

The `with-page-transform` macro can be used to establish a page transform which controls scaling by mapping a rectangular region of the document to the printable area of the page. By specifying a large rectangle, you can get finer granularity in the drawing. Any number of invocations of `with-page-transform` may occur during the printing of a page. For instance, it may be convenient to use a different page transform when printing headers and footers to the page from that used when printing the main body of the page.

A helper function, `get-page-area`, is provided to simplify the calculation of suitable rectangles for use with `with-page-transform`. It calculates the width and height of the rectangle in the user's coordinate space that correspond to one printable page, based on the logical resolution of the user's coordinate space in dpi.

For more specific control over the page transform, the printer metrics can be queried using `get-printer-metrics` and the various printer-metrics accessors such as `printer-metrics-height`.

Margins and the printable area can be set using `set-printer-metrics`.

There is an example in `examples/capi/printing/fit-to-page.lisp`.

14.8 Other printing functions

On GTK+ and Motif, printers can be added, removed and configured via `printer-configuration-dialog`. On Microsoft Windows, the equivalent functionality is on the Printer Control Panel. On Cocoa, printers are configured via the System Preferences.

A simple printing API is available via `simple-print-port`, which prints the contents of an `output-pane` to a printer.

The Hardcopy API also provides a means of printing plain text to a printer. The functions `print-text`, `print-file` and `print-editor-buffer` can be used for this.

Index

A

abort-dialog function 110, 111
:accelerator initarg 79
:action-callback initarg 38, 41, 45
:alternative initarg 79
apply-in-pane-process function 11, 65
apropos-color-alias-names function 147
apropos-color-names function 146
apropos-color-spec-names function 147
augment-font-description function 137

B

:best-height initarg 120
:best-width initarg 120
boole function 135
break gesture 14
:buffer-name initarg 25
button panels
 orientation 33
 prompting with 105–106
button-panel class 32
buttons
 check 27
 push 26
 radio 28
:buttons initarg 24

C

:callback initarg 24
callbacks

 description of 8
 general properties 45
 graph panes 41
 in interfaces 94–97
 used for choices 37–38
 using callback functions 17
:callback-type initarg 45, 111
call-editor function 95
CAPI
 basic objects 3
 description of 1–3
 linking code into 8
 menu hierarchy 76
 using the 6
check button panels 33
check buttons 27
check-button class 27, 32
check-button-panel class 32, 33, 38, 43
:children-function initarg 40
choice class 31
:choice-class initarg 105
choice-interaction accessor 44
choices 31–46
 callbacks available 45
 description of 31–46
 general properties 43–46
 relationship to menus 43
choice-selected-item accessor 44
choice-selected-items accessor 44
choice-selection accessor 39, 44
classes
 collections 32
 creating your own 117–129
clear-external-image-conversions
 function 141
CLUE 2

- CLX 2
- collection class 31
- collection-items accessor 68
- collections
 - description of 31
- collector panes 26
- collector-pane class 26
- colors
 - prompting for 107
- colors= function 150
- color-spec-model function 148
- color-spec-p function 148
- column-layout class 33, 48, 90
- column-layout class 49
- combo box 42
- combo boxes 42
- complete-in-place function 116
- confirm-yes-or-no function 100
- contain function 7, 11, 122
- convert-color function 143, 150
- convert-external-image function 141
- convert-to-screen function 12, 13, 14, 15
- :create-callback initarg 65, 142
- creating menus 71
- creating submenus 72
- current-printer function 153

D

- :data callback type 45
- :data initarg 8, 27
- :data-interface callback type 45
- default settings
 - selections 39
- :default-initargs class option 87, 90
- defclass macro 85, 87, 120
- define-color-alias function 148
- define-color-models macro 151
- define-interface macro 85
 - arguments supplied to 87
- defpackage function 6
- delete-color-translation function 148, 151
- :description initarg 48
- description of the CAPI 1–3
- destroy generic function 69
- dialogs
 - creating your own 110–114
 - description of 99–114
 - in front 109
 - modal 108
 - owners 109–110

- display callback 119
- display function 7, 8, 11, 14, 15
- display panes 23
 - :display-callback initarg 142
- display-dialog function 110, 112, 115
- displaying text on screen 23
- display-message function 9, 100
- display-pane class 23, 50
- dividers 62
- document-frame class 63
- double buffering 131
- draw-circle function 119
- draw-image function 139
- drawn-pinboard-object class 122, 126, 133
- draw-point function 121
- drop-down list box 42

E

- editor panes 24
- editor-pane class 25, 50, 95, 116
 - subclasses 25
- editor-pane-blink-rate function 25
- editor-pane-text accessor 68, 95
- :element callback type 45
- elements
 - creating your own 117–129
 - generic properties of 17–18
- element-widget-name accessor 12
- :enabled initarg 25, 27
- :enabled-function initarg 80
- :enabled-function-for-dialog initarg 81
- ensure-color function 150
- ensure-model-color function 150
- :evaluate keyword argument 108
- event handlers 120–122, ??–122
- execute-with-interface function 11, 65
- execute-with-interface-if-alive function 11, 65
- exit-confirmer function 113
- exit-dialog function 110, 111, 113
- :extend-callback initarg 38, 41, 45
- extended selection
 - specifying 44
 - using on diferent platforms 44
- :extended-selection interaction style 36–37, 44
- extension gesture 37
- external constraints 55
- external image
 - dimensions 142

- from displayed window 143
- from on-screen window 143
- width and height 142

`external-image` class 139

`externalize-image` function 141

`:external-max-height` initag 55

`:external-max-width` initag 55

`:external-min-height` initag 55

`:external-min-width` initag 55

F

`:file-completion` initag 115

files

- prompting for 106–108

`:filter` initag 39

`find-best-font` function 137

`find-matching-fonts` function 137

`:font` initag 18

fonts 18

- attributes 137

- lookup 138

- prompting for 107

frame 22

`free-image` function 139, 141

`free-image-access` function 143

functions

- sample 17

G

generic properties of elements 17–18

geometry of layouts, specifying 54–58

`get-all-color-names` function 147

`get-color-alias-translation` function 148

`get-color-spec` function 147

`get-page-area` function 156

`get-printer-metrics` function 156

graph panes

- callbacks 41

graphics

- creating permanent displays 119

- displaying 117–120

graphics ports 131

- drawing functions 135

- `pixmap` 136

graphics state 132

graphics state parameters 133

`graph-pane` class 39

- implementation of 123

`grid-layout` class 20, 51

`groupbox` 22

GTK+

- resources 12

H

hardcopy API 153–156

hierarchy of menus 76

hints 54

`:horizontal-scroll` initag 18, 49

I

`image` class 138

`image-access-pixel` function 143

`image-access-transfer-from-image` function 143

`image-access-transfer-to-image` function 143

`:image-function` initag 42, 81

`image-height` accessor 142

`image-width` accessor 142

`:initial-value` initag 102

`:in-place-completion-function` initag 115

`:in-place-filter` initag 116

`:input-model` initag 121

integers

- prompting for 102–103

interaction

- general properties 43

- in lists 36

`:interaction` initag 36, 43, 73, 104

interactive streams 26

`interactive-stream` class 26

`:interface` callback type 45

`interface` class 3, 21, 29, 85

`interface-customize-toolbar` function 29

`interface-default-toolbar-states` function 29

`interface-display` generic function 66

`interface-extend-title` generic function 21

interfaces

- defining 85–97

- description of 85

- layouts, specifying 88

- menus, specifying 91–94

- panes, specifying 88

- title, specifying 87

`interface-title` accessor 21

`interface-toolbar-state` function 29

`interface-update-toolbar` function 29

- internal constraints 55
 - :internal-max-height initarg 55
 - :internal-max-width initarg 55
 - :internal-min-height initarg 55
 - :internal-min-width initarg 55
- item-pinboard-object class 122
- :items initarg 35, 71, 75

K

- key press 120–122

L

- :layout-class initarg 33
- layout-description accessor 68
- layouts
 - combining different 52–54
 - description of 47–58
 - introduction to 7
 - specifying geometry 54–58
 - specifying size of panes in 49
- :layouts interface option 85
- letters
 - underlined in menus etc 19
- Lisp forms
 - prompting for 108
- list function 48
- list items, specifying 35
- list panels 34
- list-all-font-names function 137
- listener panes 26
- listener-pane class 26, 30
- list-panel class 20, 34
- lists
 - actions in 37
 - deselection in 37
 - extended selection in 36
 - extended selections 36
 - interaction in 36
 - multiple selection in 36
 - prompting with 103–106
 - retraction in 37
 - single selection in 36
- load-color-database function 151
- load-icon-image function 139, 141
- load-image function 141

M

- make-container function 111
- make-graphics-state function 133
- make-hsv function 147
- make-image-access function 142

- make-image-from-port function 141
- make-instance function 5
- make-menu-for-pane function 82
- make-pane-popup-menu generic function 82
- make-rgb function 147
- make-sub-image function 141
 - :max keyword argument 102
- max-height 56
- max-width 56
- MDI 63
- menu class 3, 71, 81
 - :menu-bar interface option 85, 92
 - :menu-bar-items initarg 72
- menu-component class 72
- menu-component class 3
- menu-item class 3, 75, 81
- menus
 - components 43
 - creating 71
 - creating submenus 72
 - description of 71–81
 - disabling items in 80–81
 - grouping items together 72–75
 - individual items in 75–76
 - menu hierarchy 76
 - nesting 72
 - specifying alternative items 79
- :menus interface option 85, 92
- merge-font-descriptions function 137
- Microsoft Windows
 - Multiple-Document Interface 63
 - themes 12
- :min keyword argument 102
- min-height 56
- min-width 56
- :mnemonic initarg 20, 28
- mnemonics 19
 - in a button-panel 34
 - in menus 78
- :mnemonics initarg 34
- :mnemonic-text initarg 28
- :mnemonic-title initarg 20
- modal dialogs 108
- Motif
 - resources 13
- multi-line-text-input-pane class 24
- :multiple-selection interaction style 36, 43, 75

N

- :none callback type 45

`:no-selection` interaction style 43, 44

O

off screen 131
off-screen 131
offscreen 131
`:ok-check` keyword argument 102, 108, 114
on screen 131
on-screen 131
onscreen 131
option panes 42
`option-pane` class 20, 42
organizing panes 48
`output-pane` class 82, 117, 146

P

`page-setup-dialog` function 153
`:pane-args` initarg 106
panel
 button layout 33
`pane-layout` accessor 67
panels
 check button 33
 list 34
 push button 32
 radio button 33
`:pane-menu` initarg 82
`pane-popup-menu-items` generic function 82
panes
 accessing 89
 collector 26
 creating your own 117–129
 default title position 22
 display 23
 editor 24
 finding 89
 graphs 39
 listener 26
 lookup 89
 option 42
 organizing 48
 sizing 49
 text input 23
 title 21
`:panes` interface option 85
`pane-supports-menus-with-images` function 82
`:pathname` keyword argument 107
pinboard

 buffered display 123
 double buffering 123
 flickering 123
pinboard objects 122
 creating your own 126–129
`pinboard-layout` class 52, 122, 123
`pinboard-object` class 122
`pixblt` function 131
`popup-confirmer` function 112, 113, 115
portable font descriptions 136–137
`print` function 31
`print-dialog` function 110, 153
`print-editor-buffer` function 156
`printer-configuration-dialog` function 156
`print-file` function 156
`:print-function` initarg 31
`print-text` function 156
`prompt-for-color` function 107
`prompt-for-confirmation` function 101
`prompt-for-directory` function 107
`prompt-for-file` function 106, 110
`prompt-for-font` function 107
`prompt-for-form` function 108
`prompt-for-integer` function 102, 113
`prompt-for-number` function 103
`prompt-for-string` function 101, 110, 112
`prompt-for-symbol` function 108
`prompt-with-list` function 103
`prompt-with-list-non-focus` function 116
push button panels
 creating 32
push buttons 26
`push-button` class 7, 27, 32
`push-button-panel` class 32

R

radio button panels
 creating 33
radio buttons 28
`radio-button` class 28
`radio-button-panel` class 32, 33, 43
`read-and-convert-external-image` function 141
`read-color-db` function 151
`:reader` slot option 89
`redisplay-interface` function 114
`register-image-translation` function 141
Resources
 GTK+ 12

- X11/Motif 13
- `:retract-callback` `initarg` 27, 37, 41, 45
- `row-layout` class 33, 48, 49, 90

S

- scroll bars
 - programmatic control 67
 - specifying 18
- `scroll` generic function 67
- `scroll-if-not-visible-p` accessor 67
- `:selected` `initarg` 28
- `:selected-item` `initarg` 42, 44
- `:selected-items` `initarg` 44
- selection gesture 37
- `:selection` `initarg` 39, 44
- `:selection-callback` `initarg` 32, 35, 37, 41, 45, 96
- selections 36–39
 - default settings 39
 - extending 36
 - general properties 44
 - specifying multiple 43
- separators 62
- `set-application-themed` function 12
- `set-default-interface-prefix-suffix` function 21
- `setf` function 22, 27
- `set-graphics-state` function 133
- `set-hint-table` function 58
- `set-printer-metrics` function 156
- `set-printer-options` function 153
- `set-top-level-interface-geometry` function 66
- `simple-print-port` function 156
- single selection
 - specifying 43
- `:single-selection` interaction style 36, 43, 74
- slot 5
- `slot-value` function 5
- streams
 - interactive 26
- strings
 - prompting for 101
- subclasses
 - finding 21
- subclasses, finding 21
- symbols
 - prompting for 108

T

- text
 - displaying 22
 - displaying on screen 23
 - editing 22
 - entering 22
- `:text` `initarg` 18, 23, 24, 25
- text input panes 23
- `text-input-pane` class 20, 24, 115
- `text-input-pane-in-place-complete` function 116
- `:title` `initarg` 21, 87
- title panes 21
- `titled-object` class 20
- `titled-pane-title` accessor 95
- `:title-font` `initarg` 22
- `title-pane` class 20, 146
- `:title-position` `initarg` 22, 49
- titles
 - changing 22, 97
 - changing interactively 22
 - for elements 22
 - for interfaces 21, 97
 - for windows 21, 97
 - specifying 20, 20–22
 - specifying directly 21
- toolbar
 - customize 29
 - folding 29
- toolbar buttons 29
- toolbars 29
- `top-level-interface-display-state` function 69

U

- `unconvert-color` function 143
- underlined letters 19
- user input 99–114
- using callback functions 17
- using the CAPI 6

V

- `:value-function` keyword argument 113
- values
 - prompting for 101–108
- `:vertical-scroll` `initarg` 18, 49
- visible constraints 55
- `:visible-max-height` `initarg` 55
- `:visible-max-width` `initarg` 55
- `:visible-min-height` `initarg` 55
- `:visible-min-width` `initarg` 55

W

- window titles 21, 97
- window-modal dialogs 108
- Windows themes 12
- Windows XP themes 12
- `with-dialog-results` macro 109
- `with-document-pages` macro 155
- `with-external-metafile` macro 131
- `with-graphics-state` macro 133
- `with-page` macro 155
- `with-page-transform` macro 156
- `with-pixmap-graphics-port` macro 131, 132, 143
- `with-print-job` macro 131, 153
- Works menu
 - in CAPI objects 6

X

- X resources
 - fallback resources 13
 - in delivered applications 13
- X11
 - resources 12, 13
- `:x-ratios` `initarg` 50

Y

- `:y-ratios` `initarg` 50